Graphics Programming Techniques

FM SECTION

HP System 45 Desktop Computer

Hewlett-Packard Calculator Products Division
P.O. Box 301, Loveland, Colorado 80537
(For World-wide Sales and Service Offices see back of manual.)

HP Computer Museum
www.hpmuseum.net

For research and education purposes only.

Table of Contents

Chapter 1: General Information

Introduction 1
CRT Plotting Specifications 2
For the Beginner 2
Graphics Resolution 2
Display Area 3
Line Generation Speed 3
Hardware Installation 4
Factory Installed Graphics 4
Updating with Graphics 4
Graphics Select Code 5
Graphics Memory 5
Definition of Terms 6
Plotting Space 6
Unit-of-Measure Modes 7
Line Generation 7
Syntax Guidelines 8
What to Expect 8

Chapter 2: Set-up Operations

Introduction 9
PLOTTER IS Statement 10
Statement Parameters 10
Default Conditions 12
PLOTTER...IS ON and IS OFF Statements 13
GCLEAR Statement 14
LIMIT Statement 15
LOCATE Statement 18
SCALE Statement 21
SHOW Statement 24
CLIP Statement 26
UNCLIP Statement 28
MSCALE Statement 29
SETGU and SETUU Statements 31
RATIO Function 32

Chapter 3: Plotting Operations

Introduction 35
PENUP Statement 35
PEN Statement 36
LINE TYPE Statement 37
PLOT Statement 39
MOVE Statement 42
DRAW Statement 43
RPLOT Statement 44
IPLOT Statement 46
PDIR Statement 47

Chapter 4: Axes and Labeling Operations

Introduction 51
AXES Statement 52
GRID Statement 55
FRAME Statement 57
LABEL and LABEL USING Statements 59
LETTER Statement 63
LORG Statement 64
CSIZE Statement 68
LDIR Statement 71

Chapter 5: Digitizing and Cursor Control

Introduction 73
POINTER Statement 75
CURSOR Statement 78
DIGITIZE Statement 80
WHERE Statement 84

Chapter 6: CRT Graphics Control Statements

Introduction 85
GRAPHICS and EXIT GRAPHICS Statements 86
DUMP GRAPHICS Statement 87
GLOAD and GSTORE Statements 88

Appendix A: Reflected Plots 93

Appendix B: External Plotter Installation

Introduction 97
9872A Plotter Installation 98
Interface Select Code 98
Plotter Address Code 99
Incremental Plotter Installation 101

Appendix C: Other Example Programs 103

Appendix D: Syntax Summary 111

Appendix E: Graphics Error Messages 119

Subject Index 121

Chapter 1
General Information

Introduction

The HP System 45 Graphics Package enables the computer to control the CRT and other plotters. This provides CRT graphic solutions or hard copy illustrations of problems solved by the computer. In addition Graphics provides a means to label a plot and draw axes with or without tic marks. Digitizing and cursor control allows you to retrieve numeric data from a plot or graphic display.

Before attempting to use Graphics and this manual you should be familar with the basic operation of the 9845A Computer. Refer to the 9845A Computer Operating and Programming Manual.

Information contained in this manual is directed specifically to the CRT as the graphics display unit and directed in general to all other external display or plotting units (e.g., HP 9872A Plotter).

The simulated CRT graphic drawings used to produce this manual were actually drawn on the HP 9872A Plotter. Therefore, the simulations may appear slightly different from the results you obtain on the CRT.

CRT Graphics Specifications

The System 45 Desktop Computer has two CRT display areas (rasters) or modes: **alphanumeric** and **graphics**. Normally the CRT is in the alphanumeric mode and without the Graphics ROM and hardware this is the only mode available to you.

When the Graphics ROM and hardware is installed, the graphics mode can be selected.

For the Beginner

Enter the following program into the computer and then press RUN. This program will frame (draw a box around) the CRT graphics display area that is available to you.

```
10   PLOTTER IS 13,"GRAPHICS"          ! SPECIFY CRT AS PLOTTER.
20   GRAPHICS                          ! SET CRT TO GRAPHICS MODE.
30   FRAME                             ! DRAW A FRAME AROUND PLOTTING AREA.
40   END                               ! END PROGRAM.
```

This example program and others like it found throughout this manual is given to provide some hands-on experience and to illustrate various statements. Each of the graphic syntaxes will be explained at the appropriate place in this manual.

Graphics Resolution

The horizontal and vertical resolution is approximately .04cm, with 560 usable dots in the horizontal direction and 455 usable dots in the vertical direction.

Display Area

The following figure shows the approximate size and location of the alphanumeric and graphics display areas.



Line Generation Speed

Straight lines are generated at about 200cm per second and curved lines are generated at somewhat slower speeds. This is because the computation time needed to produce a smooth curved line (e.g., sine function) decreases the apparent line generation speed.

Hardware Installation

The Graphics Package for the System 45 consists of:

- 98437A Plug-in Graphics ROM
- 98470A CRT Graphics Memory
- 09845-90050 Graphics Programming Techniques Manual

Factory Installed Graphics

If the 9845A Computer was ordered from the factory with the Graphic Option (Opt 300 and Opt 700), the hardware (ROM, CRT memory and error message sticker) will be already installed for you.

Updating With Graphics

If you are updating a 9845A Computer with Graphics, the hardware will have to be installed. The CRT Graphics Memory should be installed by trained HP Service personnel, because the installation procedure involves high voltage and a potentially dangerous situation associated with handling CRTs.

You can install the Graphics ROM in one of the open slots in the ROM drawer on the right side of the computer. If you are unfamiliar with the procedure for installing a ROM, refer to the System 45 Operating and Programming Manual for the proper procedure. You can also install the Graphics error message sticker on the option error pull-out card, which is located under the CRT as shown below. If you want, the service person will install the Graphics ROM and the error message sticker when the Graphics Memory is installed.

GRAPHICS

Error Messages

110 Plotter specification not recognized
111 Plotter not previously specified
112 No CRT Graphics hardware
113 Limit specification out of range

Selected Syntaxes	**Parameter type**
LIMIT [X min, X max, Y min, Y max]	Millimeters
LOCATE [X min, X max, Y min, Y max]	GDU's
SCALE X min, X max, Y min, Y max	Defines user units
AXES [X minor tic inc., Y minor tic inc.	X, Y in current units
[, X intersection, Y intersection	
[, X major tic count, Y major tic count	tic size in GDU's
[, tic size]]]]	
PLOT X, Y [, Pen Control]	X, Y in current units
	Pen Control:
	Even = lift pen
	Odd = drop pen
	Positive = change pen after motion
	Negative = change pen before motion



Graphics Select Code

The select code of the CRT graphics option is internally set at the factory to 13 and cannot be changed. If you are going to use an external plotting device (e.g., HP 9872A) refer to the appropriate interface Installation and Service Manual and the appendix of this manual for information about how to set the select code.

Graphics Memory

The Graphics ROM automatically decreases the available user memory by 92 bytes. This is generally not a concern unless you have a large program or large data base.

The 32K bytes of Read Write Memory (CRT Graphics Memory) that is installed in the CRT does not increase the available user memory, it only provides a storage location for the data that produces the graphics display on the CRT. When using an external display or plotting device, this memory space is not used.

Definition of Terms

Throughout this manual some new terms and mnemonics are used to describe a particular operation or syntax. In many cases a term or mnemonic is used before a complete definition can be given. For these cases, the following provides enough information to make each term meaningful without giving complete details of each. Complete details regarding each syntax are given later in the manual. The terms and mnemonics to be defined are:

- Plotting Space
  - Hard Clip Limits
  - Soft Clip Boundaries
- Unit-of-Measure Modes
  - Graphic Display Units (GDU's)
  - User Definable Units (UDU's)
  - Metric Units
  - Current Units
- Line Generation

Having a basic definition of these will aid in understanding the rest of the manual.

Plotting Space

Plotting devices have physically-imposed **hard clip limits** (e.g., the CRT's screen size and the 9872A Plotter's platen size) that restrict the pen movement. The default hard clip limits and the resulting plotting area size depends upon the plotting device used. As will be seen later, the hard clip limits can be made smaller (e.g., to fit a piece of paper) under program control, or on some plotters, manually. If this area is the default or specified size, it is still referred to as the area within the hard clip limits.

Another set of limits that further restrict the pen movement is the **soft clip boundaries**. The soft clip boundaries are defaulted to the hard clip limits, but can be altered under program control. The soft clip boundaries restrict the pen movement when the User-Definable Units (**UDU**'s) mode is set and not when the Graphic Display Units (**GDU**'s) mode is set. This capability is very useful, and is shown later in the manual.

Unit-of-Measure Modes

There are three unit-of-measure modes that can be selected under program control: **Graphic Display Units** (GDU's), **User Definable Units** (UDU's), and **Metric** units.

Most of the syntax given in this manual interpret their parameters according to the last mode set (or current units); other syntax interpret their parameters in one of the three modes.

GDU's are defined as being one percent of the length of the shortest side of the space bounded by the hard-clip limits. Thus, the short side is 100 GDU's long and the length of the long side is 100 times the ratio of the long to short dimensions. This unit mode allows plotter space access on a percent of full scale basis.

UDU's are the units defined by the user's application. Functions which operate in UDU mode can have values for the arguments directly in user units. Several functions are provided to allow you to define the units of measure for a particular application.

The unit of measure in the **Metric** unit mode is the millimetre. This mode is implemented as a special case of UDU's, such that plotting statements executed in this mode will put a line on the plotting area physically scaled to millimetres. This mode is useful where correspondence with physically measurable objects is desirable, as in drafting applications. In some devices (e.g., the CRT), METRIC units are only approximate.

The term **current units** refers to the last unit of measure mode that was set.

Line Generation

Line Generation refers to the process of producing a line on the CRT, which is similar to drawing a line with a pen on an external plotter. Unlike the external plotter, the CRT has no actual pen. The CRT does have a point, referred to as "the pen", which when moved produces a line if line generation is turned on (pen down). If line generation is turned off (pen up) no line is produced, but the point moves.

Syntax Guidelines

The following general conventions apply to the syntax which appear in this manual.

Dot Matrix - All items appearing in Dot Matrix must be entered as shown.

[ ] - All items in square brackets are optional and explained in the following text.

What to Expect

The rest of this manual is divided into chapters that describe the syntaxes that perform similar or related operations. The related operations are:

- Set Up
- Plotting
- Axes and Labeling
- Digitizing and Cursor Control
- CRT Graphics Control

In each of the chapters example programs are given to demonstrate a particular point and can be used by you to gain hands-on experience.

Chapter 2

Set-up Operations

Introduction

A program written to plot or draw lines on a plotting device must include some set-up operations to initialize the plotter and define the plotting area. However, it is not necessary to use each statement in this chapter to initialize a plotting device. The statements given in this chapter provide sufficient flexibility to accommodate most plotting applications.

If you are going to use the CRT as the plotting device, refer to GRAPHICS and EXIT-GRAPHICS statements in Chapter 6. The GRAPHICS statement has to be executed if you want to see the CRT graphics display.

Typical set-up operations are shown by the following groups of statements.

```
10   PLOTTER IS 13,"GRAPHICS"
20   GRAPHICS
30   SCALE -10,10,-10,10
40   AXES 1,1,0,0,5,5,5
50   FRAME
60   END
```

```
10   GCLEAR
20   GRAPHICS
30   SCALE -10,10,-10,10
40   AXES 1,1,-10,-10,5,5,5
50   FRAME
60   END
```

Each of these statements is covered individually in the following pages.

PLOTTER IS Statement

The PLOTTER IS statement is used to specify the select code and the type of plotting device to which successive plotter operations is directed.

Syntax:

PLOTTER IS [select code [, HP-IB device address],] plotter id string
[, step size [, # of pens, pen offset, incremental plotter id]]

Example:

```
10   PLOTTER IS 13,"GRAPHICS"
20   !   OR
30   PLOTTER IS "GRAPHICS"
40   !   AND
50   PLOTTER IS 7,5,"9872A"
60   !   OR  .
70   PLOTTER IS "9872A"
80   !   AND
90   PLOTTER IS 5,"INCREMENTAL"
100  !   OR
110  PLOTTER IS 5,"INCREMENTAL",254,4,500,2
120  !   OR
130  PLOTTER IS "INCREMENTAL"
```

Statement Parameters

The select code of the CRT (when used as a plotter) is 13 and is used for most examples in this manual.

The plotter id string (string variable or constant) specifies the type of plotting device, and therefore, the proper I/O operations which drive the plotter using the specified select code. External plotting devices (e.g. HP 9872A Plotter) set to a select code other than 13 can be connected to the computer. The plotter id string should specify the type of plotting device. The three plotter id strings and default select codes are:

"GRAPHICS"	13
"9872A"	7,5 (All HP-IB Plotters)
"INCREMENTAL"	5

CRT

The plotter id string, "GRAPHICS", specify the CRT as the display device.

9872A

The plotter id string, "9872A", specifies the HP 9872A as the plotter which must be connected to the computer via the HP 98034A HP-IB Interface. Refer to the appendix and the 98034A Installation and Service Manual (98034-90000) for hookup information.

Incremental

The plotter id string, "INCREMENTAL", specifies an incremental plotter that must be connected to the computer via the HP 98040A Incremental Plotter Interface. Refer to the appendix and the 98040A Installation and Service Manual (98040-90000) for hookup information.

The last four optional parameters of the PLOTTER IS statement provide a means for you to describe the features of the incremental plotter to the computer, so that the plotter can be driven properly. If you are not going to use an incremental plotter, do not use the last four optional parameters.

The default values for these parameters are:

Parameter	Default value
step size	.254 mm
# of pens	1
pen offset	0 mm
incremental plotter id	1

The step size parameter must be specified if the other three parameters are to be specified. The assigned value of the step size parameter is plotting device dependent and must be obtained from the plotter's documentation.

If one of the last three parameters are to be specified, they all must be specified.

The number of pens parameter is used to tell the computer the number of available pens on the plotter.

The pen offset is used to specify the center-to-center distance between pens on multi-pen devices. If the plotter has only one pen, zero offset should be specified.

The plotter id parameter is used to specify one of several known incremental plotters that will operate with the computer and the 98040A Interface. This parameter can also be used to tell the computer whether the offset is in the X or the Y direction. For more information regarding these last four parameters refer to HP 98040A Incremental Plotter Interface Installation and Service Manual.

Default Conditions

The PLOTTER IS statement sets the following default conditions when executed.

1. Activates the plotting device specified (PLOTTER... IS ON).
2. Sets UDU's (User Definable Units) as current units.
3. Sets UDU's (User Definable Units) equal to GDU's.
4. Reads hardware-set hard clip limits from the plotting device (LIMIT).
5. Defaults the CLIP and LOCATE boundaries to the hard clip limits.
6. Clears the graphics display (GCLEAR).
7. Selects pen one (PEN 1).
8. Selects line type one and the repeat length to four (LINE TYPE 1, 4).
9. Selects standard character size (CSIZE 9/15, 15/4.54).
10. Selects label origin one (LORG 1).
11. Sets label direction as left-to-right (LDIR 0).
12. Clears any error conditions.
13. Clears character count from any previous LABEL statement.

Executing the LIMIT statement sets all of the previous default conditions except the GCLEAR (default condition number 6).

The default conditions can be altered as necessary by executing one or more of the plotter statements with the proper parameters from the keyboard or from within a program.

PLOTTER...IS ON or IS OFF Statements

The PLOTTER...IS ON or IS OFF statement activates (ON) or deactivates (OFF) the plotting device specified by the select code. When activated a plotting device will respond to all of the plotting statements and when deactivated the plotting device will not respond. Executing a PLOTTER...IS ON for one select code will automatically deactivate all other plotters. If all plotters are deactivated, the plotting statements are executed but are not directed to any plotter.

Syntax:

PLOTTER select code [, HP-IB device address] IS ON

This statement activates the specified plotting device and deactivates all others. That is, all plotting statements are directed only to the specified device. This statement DOES NOT set the default conditions described under the PLOTTER IS statement.

Syntax:

PLOTTER select code [, HP-IB device address] IS OFF

This statement deactivates the plotter set to the specified select code. All plotting statements are executed but the specified plotter does not respond.

Example

```
  10   !
  20   !
• 30   PLOTTER 7,5 IS ON                  ! ACTIVATES HP-IB PLOTTER, DEVICE 5.
  40   !
• 50   PLOTTER 7,5 IS OFF                 ! DEACTIVATES HP-IB PLOTTER, DEVICE 5.
  60   !
  70   !
• 80   PLOTTER 13 IS ON                   ! ACTIVATES THE CRT AS PLOTTER.
  90   !
• 100  PLOTTER 13 IS OFF                  ! DEACTIVATES THE CRT AS PLOTTER.
  110  !
  120  !
  130  ! DO NOT RUN THIS PROGRAM, THESE ARE ONLY EXAMPLE PROGRAM LINES.
```

GCLEAR Statement

The GCLEAR statement clears the graphics memory and the CRT of any previously plotted data. The GCLEAR statement does not reset or set the default parameters for any other statement.

Syntax:

GCLEAR [paper advance]

The PLOTTER IS statement should normally be used between plots, rather than GCLEAR, since GCLEAR does not restore the default conditions.

On plotting devices with roll paper the optional parameter allows you to advance the paper by a specified amount in millimetres.

On fixed paper plotters (e.g., HP 9872A) and on the CRT the paper advance parameter is ignored.

Example

```
  10   !
  20   !
• 30   GCLEAR 2540                          ! ADVANCE PAPER 2540mm (10 INCHES).
  40   !
• 50   GCLEAR                               ! CLEAR CRT, IF IT IS CURRENT PLOTTER.
  60   !
  70   !
  80   ! DO NOT RUN THIS PROGRAM, THESE ARE ONLY EXAMPLE PROGRAM LINES.
```

LIMIT Statement

The LIMIT statement allows you to define the "plotting area" as a subset of the physical plotting area by setting the hard clip limits. In effect you are describing the size and location of a piece of paper or a specified area of the CRT to the computer. The computer is NOT allowed to move the pen outside the specified area without executing a new LIMIT statement with larger values for the parameters.

This statement overrides any previously set or defaulted hard or soft clip values. If the parameters are specified outside the physical limits of the plotting device, error 113 is generated.

If the LIMIT statement is not executed the hard clip limits default to the size of the physical plotting area. On the HP 9872A Plotter, the default hard clip area is somewhat smaller than the physical plotting area.

Again, nothing can be labelled or drawn outside the specified or default hard clip limits.

Syntax:

LIMIT [X_{min} , X_{max} , Y_{min} , Y_{max}]

The first two parameters normally specify the left and right limits and the second two parameters normally specify the lower and upper limits of the plotting area.

Because you are describing a physical piece of paper or plotting area to the computer, the units must be actual units of measure. For this syntax the units of measure are **always in millimetres** and normally the origin (point 0,0) is at the lower-left physical limits of the plotting device. The default upper-right limit values for the CRT are:

X_{max} = 184.47mm
Y_{max} = 149.82mm

Other plotting devices have their own default upper limit values.

Stated above, is the normal sequence of parameters, but the first two parameters can be exchanged to produce a reflected plot across the Y axes. Likewise the last two parameters can be exchanged to produce a reflected plot across the X axes. Both the first two parameters and second two parameters can be exchanged, at the same time, to produce a reflected plot across the origin (point 0,0). Refer to the appendix for more information on reflected plots.

Executing the LIMIT statement without parameters allows you to digitize two opposite corners to describe the plotting area. For further details about inputting the parameters by digitizing the corners, refer to the DIGITIZE Statement in Chapter 5.

The LIMIT statement is the software equivalent of manually setting the graph limits (keys P1 and P2) on a 9872A Plotter.

Example

```
10   ! THIS PROGRAM DEMONSTRATES THE USE OF THE LIMIT STATEMENT.
20   ! NOTE: LINES OR LABELS CAN NOT BE DRAWN OUTSIDE THE HARD CLIP LIMITS
30   ! UNLESS THE LIMITS ARE CHANGED BY EXECUTING ANOTHER LIMIT STATEMENT.
40   !
50   PLOTTER IS 13,"GRAPHICS"          ! SPECIFY THE CRT AS PLOTTER.
60   GRAPHICS                          ! SET THE GRAPHICS MODE.
70   FRAME                             ! FRAME THE MAX PLOTTING AREA FOR YOUR REF.
80   !
90   LIMIT 15,15+150,10,10+100         ! SPECIFY A 150mm X 100mm  PLOTTING AREA
100                                    ! THAT IS OFFSET FROM THE PLOTTERS LOWER
110                                    ! LEFT PHYSICAL BOUNDS.  THE SPECIFIED
120                                    ! OFFSET FOR THIS EXAMPLE IS 10mm UP
130                                    ! AND 15mm TO THE RIGHT. REMEMBER,
140                                    ! UNITS ARE ALWAYS IN MILLIMETRES.
150  !
160  LINE TYPE 3                       ! SPECIFY DASHED LINE FOR FRAME.
170  FRAME                             ! FRAME THE SPECIFIED PLOTTING AREA.
180  END
```

When the HP 9872A is specified is the PLOTTER IS statement and the LIMIT statement is executed without parameters, the current P1 and P2 coordinates are input as the LIMIT X_{min}, Y_{min} and X_{max}, Y_{max} points when the [ENTER] key on the plotter is pressed. These coordinates become the default hard clip limits for all subsequent programs until a new LIMIT statement is executed.

If you are running many plots on the 9872A you may wish to initialize the 9872A before executing the LIMIT statement. After executing the LIMIT statement you would then need to specify the 9872A as the plotter by re-executing a PLOTTER IS statement.


Physical Plotting Area
Specified Plotting Area
Set by LIMIT
(hard clip)
100mm
10mm
15mm
150mm

LOCATE Statement

The LOCATE statement allows you to define an area within the hard clip limits onto which the SCALE statement maps or the SHOW statement fills. In effect, you are defining an equal or smaller area (e.g., a preprinted grid on graph paper) within the plotting area (or hard clip limits). Remember, with the LIMIT statement you can define the plotting area smaller than the physical limits of the plotting device.

All lines drawn while in the UDU's mode and after executing the LOCATE statement will not extend beyond the soft clip boundaries. The pen can be positioned between the soft clip boundaries and the hard clip limits to draw labels, but plotted lines will not extend into this area. Plotting in GDU's, you can always draw lines anywhere within the plotting area. The LOCATE statement defines and resets the default soft clip boundaries.

Syntax:

LOCATE [X_{min}, X_{max}, Y_{min}, Y_{max}]

The units are **always** in Graphic Display Units (GDU's) with the origin (GDU point 0,0) normally at the lower-left corner of the plotting area (or as defined by the LIMIT statement).

The length of a GDU is equal to 1/100 (one percent) of the length of the short side of the plotting area. Therefore, the length of the short side is always 100 GDU's. The length of the long side of the plotting area is something greater than 100 depending on the plotting area aspect ratio. Note: the aspect ratio can be found with the RATIO function as shown later.

The first two parameters specify the left and right boundary limits and the last two parameters specify the lower and upper boundary limits. Like the LIMIT statement, the parameters can be exchanged to reflect the plot (refer to the LIMIT statement and Reflected Plots in the appendix).

The SCALE and SHOW statements assign user definable units of measure (UDU's) to the area specified by the LOCATE statement. The SCALE and SHOW statements act on this area in two very different ways as can be seen in the next two sections of this manual.

Executing the LOCATE statement without parameters allows you to specify the area by digitizing two opposite corners (normally the lower-left and upper-right corners). Refer to the DIGITIZE Statement in Chapter 5.

The default soft clip limits can be overridden by the CLIP statement.

The default physical length of the short side of all plotting devices is 100 GDU's but the default length of the long side is device dependent. For the CRT, the length of the long side is about 123.13 GDU's, which is the length/height aspect ratio times the height of 100 GDU's (559/454 x 100 = 123.13).

Example

```
10   ! THIS PROGRAM DEMONSTRATES THE USE OF THE LOCATE STATEMENT.
20   ! NOTE: WHILE IN THE UDU'S MODE, PLOTTED LINES CAN NOT BE DRAWN OUTSIDE THE
30   ! "LOCATE" BOUNDARIES USING THE VARIOUS PLOTTING STATEMENTS, BUT LABELS CAN
40   ! BE POSITIONED USING THE PLOTTING STATEMENTS ANYWHERE WITHIN THE "LIMIT"
50   ! HARD CLIP LIMITS OR WHILE IN THE GDU'S MODE.
60   !
70   PLOTTER IS 13,"GRAPHICS"          ! SPECIFY THE CRT AS PLOTTER.
80   GRAPHICS                          ! SET THE GRAPHICS MODE.
90   FRAME                             ! FRAME THE MAX PLOTTING AREA FOR YOUR REF.
100  !
110  LIMIT 15,15+150,10,10+100         ! SPECIFY THE PLOTTING AREA.
120  LINE TYPE 3                       ! SPECIFY DASHED LINE TYPE FOR FRAME.
130  FRAME                             ! FRAME THE SPECIFIED PLOTTING AREA.
140  !
150  LOCATE 10,50,30,90                ! "LOCATE" AN AREA WITHIN THE PLOTTING
160                                    ! AREA. AN AREA THAT YOU CAN ASSIGN
170                                    ! UNITS OF MEASURE (SEE SCALE AND SHOW).
180                                    ! REMEMBER, PARAMETERS ARE ALWAYS IN GDU'S.
190                                    ! AND GDU'S ARE A PERCENT OF THE SHORT
200                                    ! SIDE OF THE PLOTTING AREA SET BY THE
210                                    ! "LIMIT" STATEMENT.
220  !
230  LINE TYPE 4                       ! SPECIFY DASH-DOT LINE FOR FRAME.
240  FRAME                             ! FRAME THE LOCATE AREA.
250  END
```


Physical Plotting Area
Specified Plotting Area
Set by LIMIT
(hard clip)
Plotting Area
Set by LOCATE
(Default soft clip)
90 GDU's
30 GDU's
50 GDU's
10 GDU's
Short side is always 100 GDU's long

SCALE Statement

The SCALE statement defines minimum and maximum values of X and Y for the plotting area as specified by the LOCATE statement. This allows you to specify your own units for plotting.

Syntax:

SCALE X_{min}, X_{max}, Y_{min}, Y_{max}

This statement automatically sets the User Definable Units (UDU's) mode.

The first two parameters specify the values represented by the left and right boundary of the area specified by the LOCATE statement. The last two parameters specify the values represented by the lower and upper boundary in a similar manner.

For example, the first two parameters of the scale statement could specify the left edge of the plotting area as -20 and the right edge as 30. This has the effect of dividing the horizontal plotting distance into 50 units ($30 - (-20) = 50$). The last two parameters could specify different values and therefore a different scale or units for the vertical direction. These units can be used to represent distance, volume, time or whatever units your specific problem requires. The scaling factors for the X and Y directions are completely independent of each other. Thus, plots are stretched or shrunk independently in the X and Y direction to fit the plotting area (anisotropic scaling). Note that this is not the case with the SHOW statement.

The SCALE statement can be used to place the origin (point 0,0) on or off of the plotting area.

For example: if you wanted to plot the average annual rainfall at a weather station for a 10 year period, the SCALE statement might look like this:

```
SCALE 1966, 1976, 0, 20
```

where the left edge represents year 1966 and the right edge represents 1976. Rainfall would be plotted in the Y direction in volume units (e.g. inches). This allows the data to be plotted in years and volume units (e.g. 1976, 7) directly on the plotter.

Examples

```
10   ! THIS PROGRAM DEMONSTRATES HOW THE SCALE STATEMENT IS USED TO DEFINE
20   ! USER UNITS. IN THIS EXAMPLE; THE X AXIS = YEAR 1966 TO YEAR 1976 AND
30   ! THE Y AXIS = 0 INCHES TO 20 INCHES OF RAIN.
40   !
50   PLOTTER IS 13,"GRAPHICS"          ! SPECIFY THE CRT AS PLOTTER.
60   GRAPHICS                          ! PUTS THE CRT INTO THE GRAPHICS MODE.
70   FRAME                             ! FRAME THE MAX DEFAULT PLOTTING AREA.
80   LOCATE 40,100,20,70               ! DEFINES THE PLOTTING AREA.
90   SCALE 1966,1976,0,20              ! DEFINES VALUES OF LIMITS OF PLOT AREA.
100  AXES 1/12,1,1966,0,12,10          ! DRAWS AXES.
110  !
120  ! ***** DRAW RANDOM DATA FOR THIS EXAMPLE
130  MOVE 1966,0                       ! MOVE TO START OF PLOT (YR 1966, 0 INCH).
140  FOR M=1966 TO 1976 STEP 1/6       ! FOR...NEXT GENERATES RANDOM DATA FOR
150  PLOT M,RND*2+7+2*SIN(M)           ! THIS EXAMPLE.
160  NEXT M                            ! DO IT AGAIN FOR NEXT POINT.
170  END
```



```
10   ! THIS PROGRAM DEMONSTRATES THE EFFECTS OF THE SCALE STATEMENT ON THE
20   ! PLOTTING AREA. NOTE THAT THE LENGTH OF A UNIT-OF-MEASURE IN THE X
30   ! AND THE Y DIRECTION ARE NOT NECESSARILY EQUAL.
40   !
50   PLOTTER IS 13,"GRAPHICS"          ! SPECIFY THE CRT AS PLOTTER.
60   GRAPHICS                          ! SET GRAPHICS MODE.
70   DEG                               ! SET DEGREES MODE.
•80  SCALE -2,2,-4,4                   ! SPICIFY USER UNITS, USING SCALE.
90   LINE TYPE 3                       ! SPECIFY DOT LINE FOR GRID.
100  GRID 1,1                          ! DRAW GRID.
110  LINE TYPE 1                       ! SPECIFY SOLID LINE FOR FRAME AND CIRCLE.
120  FRAME                             ! FRAME PLOTTING AREA.
130  !
140  !    DRAW A CIRCLE FOR THIS EXAMPLE.
150  MOVE 0,1                          ! MOVE TO START OF CIRCLE.
160  FOR Angle=0 TO 360 STEP 15        ! FOR..NEXT LOOP TO SPECIFY AN ANGLE.
170  DRAW SIN(Angle),COS(Angle)        ! DRAW CIRCLE.
180  NEXT Angle                        ! DO IT AGAIN UNTIL COMPLETE.
190  END
```



SHOW Statement

The SHOW statement defines an area that is stretched or shrunk proportionately in the X and Y directions so that the specified area will fit in the plotting area defined by default or by the LOCATE statement. The aspect ratio of the units will be forced to be equal to one (1) by the SHOW statement. That is, one length of measure in the X direction will equal one length of measure in the Y direction (isotropic scaling).

The SHOW statement would not normally be used when you want to plot in two different units of measure, as in the previous example for the SCALE statement where inches of rainfall and years were used as units of measure. But, SHOW should be used when you want to plot something like a map where one mile along the X axes should equal one mile along the Y axes.

Syntax:

SHOW X_{min}, X_{max}, Y_{min}, Y_{max}

The first two parameters specify the acceptable upper and lower bounds for the X direction, while the last two parameters specify the acceptable bounds for the Y direction.

Example

```
10   ! THIS PROGRAM DEMONSTRATES THE EFFECTS OF THE SHOW STATEMENT ON THE
20   ! PLOTTING AREA. NOTE THAT THE LENGTH OF A UNIT-OF-MEASURE IN THE X
30   ! AND THE Y DIRECTION ARE EQUAL.
40   !
50   PLOTTER IS 13,"GRAPHICS"          ! SPECIFY THE CRT AS PLOTTER.
60   GRAPHICS                          ! SET GRAPHICS MODE.
70   DEG                               ! SET DEGREES MODE.
•80  SHOW -2,2,-4,4                     ! SPECIFY USER UNITS, USING SHOW.
90   LINE TYPE 3                       ! SPECIFY DOT LINE FOR GRID.
100  GRID 1,1                          ! DRAW GRID.
110  LINE TYPE 1                       ! SPECIFY SOLID LINE FOR FRAME AND CIRCLE.
120  FRAME                             ! FRAME PLOTTING AREA.
130  !
140  !    DRAW A CIRCLE FOR THIS EXAMPLE.
150  MOVE 0,1                          ! MOVE TO START OF CIRCLE.
160  FOR Angle=0 TO 360 STEP 15        ! FOR..NEXT LOOP TO SPECIFY AN ANGLE.
170  DRAW SIN(Angle),COS(Angle)        ! DRAW CIRCLE.
180  NEXT Angle                        ! DO IT AGAIN UNTIL COMPLETE.
190  END
```



CLIP Statement

The CLIP statement redefines the soft clip boundaries. This allows the soft clip boundaries to be moved from their default location set by the PLOTTER IS, LIMIT or LOCATE statements. Soft clipping affects lines plotted in user defined units, but has no effect on lines plotted in GDU's or as labels.

Syntax:

CLIP [X_{min} , X_{max} , Y_{min} , Y_{max}]

The parameters are interpreted according to current units of measure.

The first two parameters specify the left and right boundaries and the last two parameters specify the lower and upper boundaries of the clipping area.

Lines plotted using UDU's from inside this area to points outside the clip boundary extend no farther than the boundary. Lines plotted outside the CLIP boundaries will not be drawn, (except when drawn in GDU's).

Executing the CLIP statement without parameters allows you to specify the area by digitizing two opposite corners (normally the lower-left and upper-right corners). Refer to the DIGITIZE statement in Chapter 5 and Reflected Plots in the appendix.

The soft clip limits can be turned off (by UNCLIP) or may be set beyond the hard clip limit. In either case and in the case of plotting commands in GDU's only the hard clip limit (set by the LIMIT statement or defaulted by the PLOTTER IS statement) is used.

Executing PLOTTER IS, LIMIT, LOCATE or UNCLIP will reinstate various default values including the soft clip boundaries, refer to the section about each statement.

When CLIP is used to set the soft clip boundaries such that the soft clip rectangle is partially in and partially out of the hard clip limits, UDU plotting is clipped within the intersection of these two limits.

Example

```
10   !
20   !
30   CLIP 20,80,20,90                    ! CLIP LINES DRAWN IN USER UNITS
40                                       ! AT THE SOFT CLIP BOUNDARIES SPECIFIED
50                                       ! IN CURRENT UNITS.
60   !
70   CLIP Xmin,Xmax,Ymin,Ymax            ! SPECIFY SOFT CLIP BOUNDARIES.
80   !
90   !
100  ! DO NOT RUN THIS PROGRAM, THESE ARE ONLY EXAMPLE PROGRAM LINES.
```

Plotting in GDU's



Plotting in UDU's



The above two figures show the clipping action to expect when drawing a line from point 1, to point 2, to point 3 and then to point 4. The rectangles represent clipping boundaries or limits for the associated statements (LOCATE, LIMIT and CLIP).

Notice that the line drawn in the first figure (plotted in GDU's) is clipped only by the hard clip limits, and the line drawn in the second figure (plotted in UDU's) is clipped at the intersection of the two (LOCATE and CLIP) soft clip boundaries.

UNCLIP Statement

The UNCLIP statement sets the soft clip boundaries equal to the hard clip limits. This allows you to draw lines anywhere in the plotting area defined by the LIMIT statement while in the user units mode.

Syntax:

```
UNCLIP
```

Clipping is set by executing PLOTTER IS, LIMIT, LOCATE, or CLIP statements. SCALE and SHOW do not affect the clipping limits.

Example

```
10   !
20   !
30   UNCLIP                              ! RESET THE CLIP LIMITS TO "LOCATE"
40                                       ! BOUNDARIES WITHIN THE HARD CLIP
50                                       ! LIMITS.
60   !
70   !
80   ! DO NOT RUN THIS PROGRAM, THIS IS ONLY AN EXAMPLE PROGRAM LINE.
```

MSCALE Statement

The MSCALE statement sets millimetres as user units and defines the origin. This mode is very useful where correspondence with physically measurable objects is desirable, as in drafting and mapping applications. In some devices (e.g., the CRT), metric units are only approximate.

Syntax:

MSCALE X_{offset}, Y_{offset}

MSCALE specifies that current user units are millimetres, and that the origin or reference point, that is the coordinate (0,0) in millimetre-space, is offset from the minimum hard clip corner by the specified amounts in millimetres. The specified reference point need not be inside the hard clip limits.

Example

```
 10   ! THIS PROGRAM DRAWS A METRIC RULER (SCALE) AND DEMONSTRATES HOW
 20   ! THE MSCALE, CLIP, AND FRAME STATEMENTS CAN BE VERY USEFUL.
 30   ! NOTE: DELETE THE "DUMP GRAPHICS" PROGRAM LINE IF YOUR SYSTEM 45
 40   ! DOES NOT HAVE AN INTERNAL PRINTER.
 50   !
 60   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY THE CRT AS PLOTTER.
 70   GRAPHICS                            ! SET THE CRT GRAPHICS MODE.
 80   FRAME                               ! FRAME THE DEFAULT PLOTTING AREA.
 90   !
•100  MSCALE 20,50                        ! OFFSET POINT 0,0 UP 50mm & RIGHT 20mm
 110                                      ! AND SET MILLIMETRES AS CURRENT UNITS.
 120  !
 130  ! ***** DRAW WITH TIC-MARKS A 10cm SCALE.
 140  CLIP 0,100,0,10                     ! DEFINE SIZE FOR THE FRAME STATEMENT.
 150  FRAME                               ! FRAME THE METRIC SCALE PLOT.
 160  AXES 2,10,0,10,5,10,5               ! DRAW TIC MARKS (AXES) ON SCALE.
 170  !
 180  ! ***** LABEL THE SCALE.
 190  MOVE 50,3                           ! MOVE TO CENTER OF SCALE.
 200  LORG 5                              ! SPECIFY TO CENTER LABEL.
 210  LABEL USING "K";"10cm METRIC SCALE" ! LABEL SCALE.
 220  !
 230  ! ***** DUMP GRAPHICS TO THE INTERNAL PRINTER.
 240  DUMP GRAPHICS                       ! DUMP THE DATA TO THE INTERNAL PRINTER
 250  END
```


10cm (UDU's)
10cm METRIC SCALE
0,0 (UDU's)
50mm Y Offset
20mm X Offset

SETGU and SETUU Statements

Several graphic commands (as indicated) interpret their parameters using the units currently selected. These two commands provide the means to change the units mode.

Syntax:

SETGU

or

SETUU

SETGU sets or resets the graphic display units (GDU's) and SETUU sets or resets the user definable units (UDU's). UDU's are also reset by the PLOTTER IS and LIMIT statements. The default UDU's are defined to be equal to the GDU's until either a SCALE, SHOW or MSCALE statement is executed.

GDU's are primarily intended to position things on the the plotting device (i.e., the plotting area as specified by the LIMIT statement).

UDU's are defined by the most recently executed SCALE, SHOW or MSCALE statement.

Example

```
10   !
20   !
•30  SETGU                          ! SET GDU'S MODE.
40   !
•50  SETUU                          ! SET UDU'S MODE.
60   !
70   !
80   ! DO NOT RUN THIS PROGRAM, THESE ARE ONLY EXAMPLE PROGRAM LINES.
```

RATIO Function

The RATIO function returns a value equal to the ratio of the physical dimension of the hard clip limits. That is, the X dimension divided by the Y dimension.

Syntax:

```
RATIO
```

The function could be used to find the maximum length of the long side of the hard clip area. That is, multiply the length of the short side times RATIO. Remember that the length of the short side is always equal to 100 GDU's.

Example

```
Aspect=RATIO
```

The following program will always move the pen to the opposite corner of point 0, 0, (in GDU's) regardless of the size or shape of the plotting area.

Example

```
10   ! THIS PROGRAM DRAWS A LINE TO THE OPPOSITE CORNER FROM GDU POINT 0,0.
20   ! IN OTHER WORDS, GIVES YOU THE MAXIMUM VALUES OF THE PLOTTER LIMITS
30   ! IN GDU'S. NOTE: THE MINIMUM VALUE IS ALWAYS 0,0 GDU'S.
40   !
50   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY CRT AS PLOTTER.
60   GRAPHICS                              ! SET THE CRT GRAPHICS MODE.
70   LIMIT                                 ! YOU CAN SPECIFY THE PLOTTING AREA.
80                                         ! BY DIGITIZE TWO OPPOSITE CORNERS.
90   FRAME                                 ! FRAME THE PLOTTING AREA.
100  MOVE 0,0                              ! MOVE TO POINT 0,0 IN GDU'S.
110  !
120  ! ***** COMPUTE MAX & MIN VALUES (GDU'S) OF THE SIZE OF THE PLOTTING AREA.
•130 Xgdu_max=100*MAX(1,1*RATIO)
•140 Ygdu_max=100*MAX(1,1/RATIO)
150  !
160  DRAW Xgdu_max,Ygdu_max                ! DRAW A LINE TO OPPOSITE CORNER.
170  END
```


Xmax, Ymax GDU's
0,0 GDU's

Chapter 3
Plotting Operations

Introduction

The statements presented in this chapter allow you to control the pen number, line type and pen's movement, to produce lines for graphic display.

PENUP Statement

The PENUP statement lifts the pen.

Syntax:

```
PENUP
```

On plotting devices with actual pens (e.g. 9872A Plotter) this statement lifts the pen so it can be moved without drawing a line.

On the CRT this statement turns off line generation. If the pen is moved (with the pen up) its new location will not be apparent on the CRT, but the pen's coordinate values can be found by executing the WHERE statement (refer to Chapter 5).

The pen up or pen down status can be automatically controlled by using the DRAW and MOVE statements or the PLOT, RPLOT and IPLOT statements with the optional pen control parameter.

Example

```
 10   !
 20   !
•30   PENUP                              ! RAISE THE PEN OR STOP LINE GENERATION
 40   !
 50   !
 60   ! DO NOT RUN THIS PROGRAM, THIS IS ONLY AN EXAMPLE PROGRAM LINE.
```

PEN Statement

The PEN statement allows you to select any one of the four pens found on the plotting device.

Syntax:

PEN pen number

The meaningful range of the pen number parameter is minus one (−1) through plus four (+4). Non-integer pen number parameters are rounded (up or down) to the nearest integer value.

Pen number zero (0) selects a blank pen. That is, on the 9872A Plotter all pens are returned to their holders.

On external plotters that have more then one pen, the absolute value of the pen parameter is used to select one of four pens (possibly of different colors).

On the CRT, any positive pen number turns on line generation. A negative pen number selects an eraser, a line redrawn with a negative pen number will be erased along with the intersecting points of any intersecting lines. Note: due to the method used to generate lines on the CRT all of a line may not be erased if you do not erase its entire length. This is most likely to occur on other than horizontal or verical lines. The tic marks drawn by line type nine will also not be erased.

Executing a PEN statement with a negative pen number (the eraser) will reset the LINE TYPE to type one, (refer to the next section of this manual).

Example

```
 10   ! THIS PROGRAM DEMONSTRATES HOW THE PEN STATEMENT IS USED TO
 20   ! DRAW AND ERASE LINES ON THE CRT. A FRAME IS DRAWN WITH A POSITIVE
 30   ! PEN NUMBER AND THEN ERASED WITH A NEGATIVE PEN NUMBER.
 40   !
 50   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY THE CRT AS PLOTTER.
 60   GRAPHICS                            ! SET THE CRT GRAPHICS MODE.
 70   P=1                                 ! INITIALIZE P, A PEN THAT DRAWS.
 80 Loop:  !
*90   PEN P                               ! SPECIFY DRAW OR ERASE DEPENDING ON P.
 100  FRAME                               ! FRAME THE PLOTTING AREA.
 110  IF P=1 THEN WAIT 500                ! WAIT 1/2 SEC IF P=1, FOR DISP TIME.
 120  P=-P                                ! SWITCH TO OPPOSITE VALUE OF P.
 130  GOTO Loop                           ! REPEAT THE LOOP.
 140  END
```

LINE TYPE Statement

The LINE TYPE statement selects one of several solid or dashed line types and defines the length of the repeat pattern for the dashed or dotted lines.

Syntax:

LINE TYPE id number [, length]

The id number parameter (integer value 1 through 9) selects one of the nine line types for plotting (see the example below). The default line type is type one.

The length parameter specifies the approximate length of the dashed line repeat pattern in GDU's. The default length is 4 GDU's.

Line type patterns and the results obtained from the CRT are shown below.

NOTE

The line type patterns drawn by external plotters are slightly different from the line types drawn on the CRT. Refer to the results of the example program when drawn on the 9872A Plotter.

1. Solid
2. End Point
3. Dashed
4.
5.
6.
7.
8.
9. Tic Marks

Example

```
10   ! THIS PROGRAM DRAWS THE NINE AVAILABLE LINE TYPES.
20   !
30   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY THE CRT AS PLOTTER.
40   GRAPHICS                            ! SET THE CRT GRAPHICS MODE.
50   FRAME                               ! FRAME THE PLOTTING AREA.
60   SCALE 0,10,-10,0                    ! SPECIFY USER UNITS.
70   LORG 2                              ! CENTER LABELS ON LINES.
80   !
90   ! ***** LABEL AND DRAW AVAILABLE LINE TYPES.
100  FOR Line=1 TO 9                     ! SPECIFY ONE OF 9 LINES.
110  MOVE 1,-Line                        ! MOVE TO LABEL POSITION.
•120 LINE TYPE 1                         ! SPECIFY LINE TYPE FOR LABEL.
130  LABEL USING "K";Line                ! LABEL LINE.
•140 LINE TYPE Line                      ! SPECIFY LINE TYPE FOR LINE.
150  MOVE 2,-Line                        ! MOVE TO START OF LINE.
160  DRAW 8,-Line                        ! DRAW LINE.
170  NEXT Line                           ! DO IT AGAIN FOR NEXT LINE TYPE.
180  END
```



Line Types for the 9872A Plotter

*On the 9872A Plotter, line types 7 and 8 do not change the previous set line types but will light the error light on the 9872A. This error condition can be ignored. The line types for the CRT are shown on the previous page.

PLOT Statement

The PLOT statement provides absolute data plotting with optional pen control. This statement moves the pen to the absolute X,Y coordinate in current units using the current pen number and line type.

Syntax:

PLOT $X_{parameter}$, $Y_{parameter}$ [, pen control]

The X and Y parameters are interpreted according to the current units mode.

The optional pen control parameter specifies the up or down pen movement, and defaults to plus one (the pen moves and then drops). The pen control parameter is interpreted as follows.

Odd	=	Drop pen (pen down or start line generation)
Even	=	Lift pen (pen up or stop line generation)
Positive	=	Pen change after motion
Negative	=	Pen change before motion

Examples

+1 or no parameter	Move or draw then drop pen
2 or 0	Move or draw, then lift pen
−2	Lift pen, then move
−1	Drop pen, then draw

PLOT is the preferred pen moving statement for data-controlled plotting. The pen control parameter allows the data to control the pen's move or draw action, rather then by choice of a program statement as necessary with the MOVE and DRAW statements.

All plotted lines are clipped as described for the CLIP statement in Chapter 2. The pen can be moved, using the PLOT statement, outside the soft clip boundaries, but yet inside the hard clip limits to position labels (refer to the LIMIT, LOCATE and CLIP statements).

```
10   ! THIS PROGRAM DRAWS FOUR FIGURES FROM DATA FOUND IN THE DATA STATEMENTS.
20   ! THE PLOT STATEMENT ALLOWS DATA TO CONTROL THE PEN POSITION AND THE
30   ! PEN STATUS (UP OR DOWN).
40   !
50   !
60   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY CRT AS PLOTTER.
70   GRAPHICS                              ! SET GRAPHICS MODE.
80   !
90   ! ***** DEFINE MAX & MIN VALUES FOR SCALE STATEMENT.
100  Xmin=-10
110  Xmax=10
120  Ymin=-10
130  Ymax=10
140  SCALE Xmin,Xmax,Ymin,Ymax             ! DEFINE AND SPECIFY USER UNITS.
150  !
160  ! ***** DRAW AXES AND FRAME CLIP AREA.
170  AXES 1,1,0,0,5,5                      ! DRAW AXES.
180  FRAME                                 ! FRAME PLOTTING AREA.
190  !
200  !
210  ON ERROR GOTO Out_of_data             ! SPECIFY BRANCH WHEN OUT OF DATA.
220 Loop: !
230  READ Xvalue,Yvalue,Pen_control        ! READ UNTIL ERROR 36, (OUT OF DATA).
240  PLOT Xvalue,Yvalue,Pen_control        ! PLOT DATA FROM LIST WITH PEN CONTROL.
250  GOTO Loop                             ! DO IT AGAIN, UNTIL OUT OF DATA.
260  !
270  !
280 Out_of_data:!
290  IF ERRN=36 THEN GOTO 330              ! IF ERROR IS 36 (OUT OF DATA) GOTO END
300  BEEP                                  ! BEEP FOR ERROR.
310  EXIT GRAPHICS                         ! EXIT GRAPHICS TO DISPLAY ERROR.
320  DISP ERRM$                            ! DISPLAY THE ERROR MESSAGE.
330  END                                   ! END OF EXAMPLE.
340  !
350  !
360  ! ***** X, Y AND PEN CONTROL DATA FOR PLOT STATEMENT.
370  ! DATA FOR FIRST FIGURE
380  DATA 2,2,1
390  DATA 8,2,-1
400  DATA 5,8,-1
410  DATA 2,2,2
420  ! DATA FOR SECOND FIGURE
430  DATA -4,2,-2
440  DATA -4,4,-1
450  DATA -2,4,-1
460  DATA -5,9,-1
470  DATA -8,4,-1
480  DATA -6,4,-1
490  DATA -6,2,-1
500  DATA -4,2,2
510  ! DATA FOR THIRD FIGURE
520  DATA -2,-2,1
530  DATA -8,-2,-1
540  DATA -8,-8,-1
550  DATA -2,-8,-1
560  DATA -2,-2,2
570  ! DATA FOR FOURTH FIGURE
580  DATA 2,-2,1
590  DATA 8,-2,-1
600  DATA 2,-8,-1
610  DATA 8,-8,-1
620  DATA 2,-2,2
```



MOVE Statement

The MOVE statement lifts the pen and then moves to the absolute X, Y coordinate in current units. This statement provides an easy way of moving the pen without drawing a line and without regard to whether the pen is currently up or down. The pen's movement is restricted only by the hard clip limits.

Syntax:

MOVE $X_{parameter}$, $Y_{parameter}$

The X and Y parameters are interpreted according to the current units.

MOVE X, Y is equivalent to PLOT X, Y, -2 (refer to the PLOT statement).

Example

```
  10   !
  20   !
• 30   MOVE 2,5                          ! MOVE WITH PEN UP TO POINT 2,5.
  40   !
  50   !
  60   !
• 70   MOVE 25,500                       ! MOVE WITH PEN UP TO POINT 25,500.
  80   !
  90   !
  100  ! DO NOT RUN THIS PROGRAM, THESE ARE ONLY EXAMPLE PROGRAM LINES.
```

DRAW Statement

The DRAW statement drops the pen and then moves to the absolute X, Y coordinate in current units using the current pen number and line type. This statement provides an easy way of drawing a line from the current pen's location to a new location without regard as to whether the pen is currently up or down.

Syntax:

DRAW $X_{parameter}$, $Y_{parameter}$

The X and Y parameters are interpreted according to the current units.

DRAW X, Y is equivalent to PLOT X, Y, −1 (refer to the PLOT statement).

The line that is drawn will be clipped as described for the CLIP statement.

When plotting is GDU's only the hard clip limits are used. That is, drawn lines will extend no farther then the hard clip limit in any direction.

When plotting in UDU's the line will extend no farther then the soft clip boundaries in any direction.

Example

```
  10   !
  20   !
• 30   DRAW 2,5                          ! DRAW A LINE TO POINT 2,5.
  40   !
  50   !
  60   !
• 70   DRAW 25,500                       ! DRAW A LINE TO POINT 25,500.
  80   !
  90   !
  100  ! DO NOT RUN THIS PROGRAM, THESE ARE ONLY EXAMPLE PROGRAM LINES.
```

RPLOT Statement

The RPLOT statement provides relative plotting capability with pen control. The origin is assumed to be the last absolute plotted point. The line is drawn using the current pen number and line type.

Syntax:

RPLOT $X_{parameter}$, $Y_{parameter}$ [, pen control]

The RPLOT statement interprets the X and Y parameters according to current units relative to a local origin. The local origin is the last absolute plotted point resulting from one of the following statements.

PLOT	DRAW	MOVE
FRAME	AXES	GRID
LABEL	IPLOT	

The local coordinate system can be rotated about it's origin relative to the master coordinate system by means of the PDIR (plot direction) statement. That is, a figure can be drawn or repeated with the RPLOT statement at another angle by specifying the angle with the PDIR statement.

The optional pen control parameter specifies the up or down pen movement, and defaults to plus one (the pen moves and then drops). The pen control parameter is interpreted as follows.

Odd	=	Drop pen (pen down or start line generation)
Even	=	Lift pen (pen up or stop line generation)
Positive	=	Pen change after motion
Negative	=	Pen change before motion

When plotting in GDU's only the hard clip limits are used. That is, drawn lines will extend no farther then the hard clip limit in any direction.

When plotting in UDU's the line will extend no farther then the soft clip boundaries in any direction.

Example

```
10   ! THIS PROGRAM DEMONSTRATES A WAY TO DRAW PI-CHARTS USING RPLOT AND
20   ! SPECIFYING JUST THE LOCATION OF THE CENTER OF THE CIRCLE.
30   !
40   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY CRT AS PLOTTER.
```

```
 50   GRAPHICS                                  ! SET GRAPHICS MODE.
 60   DEG                                       ! SET DEGREES MODE.
 70   SHOW 0,10,-1,1                            ! SPECIFY USER UNITS.
 80   FRAME                                     ! FRAME THE PLOTTING AREA.
 90   !
 100  ! ***** DRAW THREE PI-CHARTS.
 110  FOR Circle=2 TO 8 STEP 3                  ! SPECIFY ONE OF THREE CIRCLES
 120  MOVE Circle,0                             ! MOVE TO POINT TO DRAW CIRCLE.
 130  FOR Arc=0 TO 360 STEP 15                  ! SPECIFY POINT ON CIRCLE TO DRAW ARC.
• 140  PDIR Arc                                 ! SPECIFY PLOT DIRECTION FOR "RPLOT".
• 150  RPLOT 1,0                                ! DRAW ARC.
 160  NEXT Arc                                  ! DO IT AGAIN, UNTIL COMPLETE.
 170  !
 180  ! ***** DRAW RANDOM SECTION LINES FOR THIS EXAMPLE.
 190  FOR Section=1 TO RND*5+2                  ! SPECIFY RANDOM NUMBER OF SECTIONS.
 200  MOVE Circle,0                             ! MOVE TO CENTER OF CIRCLE.
• 210  PDIR RND*360                             ! SPECIFY RANDOM DIRECTION FOR LINE.
• 220  RPLOT 1,0,-1                             ! DRAW SECTION LINE.
 230  NEXT Section                              ! DO IT AGAIN, UNTIL COMPLETE.
 240  !
 250  NEXT Circle                               ! DO NEXT CIRCLE.
 260  END
```



IPLOT Statement

The IPLOT statement provides incremental plotting capability with pen control. The origin is assumed to be the last plotted point. That is the origin is always the pen's position before this statement is executed.

Syntax:

IPLOT $X_{increment}$, $Y_{increment}$ [, pen control]

The IPLOT statement interprets the X and Y parameters according to the current units relative to a local origin. The local origin is that of the pen position before the IPLOT statement is executed (i.e. the current pen position).

The local coordinate system can be rotated about it's origin relative to the master coordinate system by means of the PDIR (plot direction) statement. That is, a figure can be drawn or repeated at another angle by specifying the angle with the PDIR statement. Refer to the PDIR statement for examples.

The optional pen control parameter specifies the up or down pen movement, and defaults to plus one (the pen moves and then drops). The pen control parameter is interpreted as follows.

Odd	=	Drop pen (pen down or start line generation)
Even	=	Lift pen (pen up or stop line generation)
Positive	=	Pen change after motion
Negative	=	Pen change before motion

When plotting in GDU's only the hard clip limits are used. That is, drawn lines will extend no farther then the hard clip limit in any direction.

When plotting in UDU's the line will extend no farther then the soft clip boundaries in any direction.

Example

```
10   !
20   !
30   IPLOT 1,3,-1                    ! PLOT ONE UNIT TO THE RIGHT AND
40                                   ! THREE UNITS UP, WITH PEN DOWN.
50   !
60   IPLOT -5,2                      ! PLOT FIVE UNITS TO THE LEFT AND
70                                   ! TWO UNITS UP, WITH CURRENT PEN
80                                   ! STATUS (UP OR DOWN) THEN DROP PEN.
90   !
100  ! DO NOT RUN THIS PROGRAM, THESE ARE ONLY EXAMPLE PROGRAM LINES.
```

PDIR Statement

The PDIR statement sets the angle of rotation for relative (RPLOT) and incremental (IPLOT) plotting.

Syntax:

PDIR angle

The angle parameter is assumed to be the counter-clockwise angle in current angle units (DEG, RAD, or GRAD) from the normal (horizontal) X axis, extending from left to right.

A figure drawn using the RPLOT or IPLOT statements can be rotated about the origin (refer to RPLOT and IPLOT) by specifying an angle with the PDIR statement.

On the CRT you can make a figure appear to move or rotate by erasing the old figure and redrawing it after executing a PDIR statement with a new angle. To erase the old figure on the CRT execute a GCLEAR statement or redraw the old figure using a negative pen number (see PEN).

Example

```
10   ! THIS PROGRAM SHOWS HOW THE PDIR STATEMENT CAN BE USED TO ROTATE
20   ! A SET OF FIXED POINTS. IN THIS CASE THE HANDS OF A CLOCK.
30   !
40   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY THE CRT AS PLOTTER.
50   GRAPHICS                            ! SET CRT GRAPHICS MODE.
60   SHOW -6,6,-6,6                      ! SPECIFY USER UNITS.
70   FRAME                               ! FRAME PLOTTING AREA.
80   DEG                                 ! SET DEGREES MODE.
90   !
100  ! ***** DRAW THE FACE OF THE CLOCK.
110  CSIZE 10                            ! SPECIFY CHARACTER SIZE.
120  LORG 5                              ! SPECIFY TO CENTER NUMBERS.
130  FOR Minute=1 TO 60                  ! FOR...NEXT TO SPECIFY MINUTE MARK.
140  MOVE 0,0                            ! MOVE TO CENTER OF CLOCK.
150  PDIR 360*(1-Minute/60)              ! PLOT DIRECTION FOR MINUTE.
160  IPLOT 0,4.5,-2                      ! MOVE TO RIM OF CLOCK.
170  IPLOT 0,.1,-1                       ! DRAW MINUTE MARK.
180  IF Minute MOD 5 THEN GOTO 220       ! IF NOT ON 5 MINUTE MARK, SKIP TO NEXT
190  IPLOT 0,.2                          ! DRAW LONGER 5 MINUTE MARK.
200  IPLOT 0,.7,-2                       ! MOVE TO CENTER OF NUMBER.
210  LABEL USING "K";Minute/5            ! LABEL WITH A NUMBER.
220  NEXT Minute                         ! DO IT AGAIN FOR NEXT MINUTE.
230  !
240  ! ***** ENTER THE CURRENT TIME.
250  EXIT GRAPHICS                       ! EXIT GRAPHICS FOR INPUT.
260  INPUT "ENTER THE HOUR",Hour         ! ENTER HOUR.
270  Hour=INT(Hour) MOD 12               ! ADJUST HOUR TO 12 HOUR CLOCK.
280  INPUT "ENTER THE MINUTES",Minute    ! ENTER MINUTE.
```

```
 290  Minute=INT(Minute) MOD 60           ! ADJUST MINUTE TO 60 MINUTE HOUR.
 300  Minute=Minute+60*Hour               ! COMBINE MINUTES AND HOURS.
 310  GRAPHICS                            ! GO BACK TO GRAPHICS.
 320  !
 330  ! ***** DRAW, ERASE AND ROTATE A BOX USING PDIR AND IPLOT STATEMENTS.
 340 Loop:!
 350  PEN -1                              ! SPECIFY ERASE (PEN -1).
 360  GOSUB Second_hand                   ! ERASE SECOND HAND.
 370  Second=(Second+1) MOD 60            ! INCREMENT SECONDS.
 380  PEN 1                               ! SPECIFY TO DRAW (PEN 1).
 390  GOSUB Second_hand                   ! DRAW SECOND HAND.
 400  IF NOT Second THEN PEN -1           ! SPECIFY ERASE IF SECOND = 0.
 410  GOSUB Minute_hand                   ! DRAW OR ERASE MINUTE HAND.
 420  GOSUB Hour_hand                     ! DRAW OR ERASE HOUR HAND.
 430  IF NOT Second THEN Minute=Minute+1  ! IF SECOND = 0, INCREMENT MINUTES.
 440  PEN 1                               ! SPECIFY DRAW.
 450  GOSUB Minute_hand                   ! DRAW MINUTE HAND.
 460  GOSUB Hour_hand                     ! DRAW HOUR HAND.
 470  WAIT 300                            ! WAIT TO ADJUST CLOCK SPEED.
 480  GOTO Loop                           ! DO IT AGAIN FOR NEXT SECOND.
 490  !
 500  ! ***** DRAW ROUTINES FOR THE SECOND, MINUTE AND HOUR HANDS.
 510  !
 520  !  DATA FOR SECOND HAND.
 530 Second_hand: !
•540  PDIR -Second*6
 550  MOVE 0,0
•560  IPLOT 0,4.4,-2
•570  IPLOT -.2,-.4,-1
•580  IPLOT .4,0
•590  IPLOT -.2,.4
 600  RETURN
 610  !
 620  !  DATA FOR MINUTE HAND.
 630 Minute_hand:!
•640  PDIR -Minute*6
 650  MOVE 0,0
•660  IPLOT -.25,3,-1
•670  IPLOT .25,1
•680  IPLOT .25,-1
•690  IPLOT -.25,-3
 700  RETURN
 710  !
 720  !  DATA FOR HOUR HAND.
 730 Hour_hand: !
•740  PDIR -Minute*6/12
 750  MOVE 0,0
•760  IPLOT -.4,2,-1
•770  IPLOT .4,1
•780  IPLOT .4,-1
•790  IPLOT -.4,-2
 800  RETURN
```


12
1
2
3
4
5
6
7
8
9
10
11

Chapter 4
Axes and Labeling Statements

Introduction

The statements presented in this chapter will allow you to place axes and labels on the plotting area. The axes can be in the form of the normal X and Y axis lines, a grid or a line around the plotting area (frame). Labels can be drawn under program control or typed in from the keyboard. You can specify the size, shape, rotation and location of the labels.

AXES Statement

The AXES statement draws a pair of axes, with optional (linearly-spaced) tic-marks.

Syntax:

AXES [$X_{tic\ spacing}$, $Y_{tic\ spacing}$ [, $X_{intersection}$, $Y_{intersection}$ [, $X_{major\ count}$, $Y_{major\ count}$ [, major-tic size]]]]

The X and Y tic-spacing parameter values are interpreted in current units. The sign of each parameter is ignored.

The AXES statement generates an X axis at $Y_{intersection}$ and a Y axis at $X_{intersection}$, assuming that these values are within the current user units clipping limits. The axes will extend across the soft clipping area. The default value for X and Y intersection is 0,0.

The X and Y major counts are unitless integer values which specify the number of minor tic intervals between major tic-marks. The tic marks are positioned along each axis such that a major tic-mark falls on the origin (where visible or not). Tic marks may or may not coincide with the edge of the clipping boundary. The sign of each major count parameter is ignored and the default value for each is one (i.e. all major tics).

The major-tic size specifies the length of the major tic-marks (end to end) and like the minor tic-marks, they are symmetric about the axes. The minor tics are one half the length of the major tics. The default size of the major tic-marks is 2 GDU's. Both the major and minor tic-marks are clipped at the clipping limits.

Axes and tic-marks are drawn using the current line type and pen number (refer to Chapter 3).

Examples

```
10   ! THIS PROGRAM DRAWS AN AXES AND FRAMES THE PLOTTING AREA.
20   !
30   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY THE CRT AS PLOTTER.
40   GRAPHICS                            ! SET THE CRT GRAPHICS MODE.
50   SCALE -10,10,-10,10                 ! SPECIFY USER UNITS.
60   !
70   AXES 1,2,0,0,5,2,5                  ! DRAW AXES, SPECIFYING:
80                                       ! ONE MINOR TIC PER UNIT IN X,
90                                       ! ONE MINOR TIC PER TWO UNITS IN Y,
100                                      ! INTERSECT AT 0,0,
110                                      ! ONE MAJOR TIC PER 5 MINOR TICS IN X,
120                                      ! ONE MAJOR TIC PER 2 MINOR TICS IN Y,
130                                      ! MAJOR TIC SIZE IS 5 GDU'S.
140  !
150  FRAME                               ! FRAME PLOTTING AREA.
160  END
```


0,0
One Unit of X
Two Units of Y

```
10   ! THIS PROGRAM DRAWS AN AXES AND FRAMES THE PLOTTING AREA.
20   !
30   GCLEAR                              ! CLEARS THE CRT OF GRAPHICS.
40   GRAPHICS                            ! SET THE CRT GRAPHICS MODE.
50   SCALE -10,10,-10,10                 ! SPECIFY USER UNITS.
60   !
70   AXES 1,2,-10,-10,5,2,5              ! DRAW AXES, SPECIFYING:
80                                       ! ONE MINOR TIC PER UNIT IN X,
90                                       ! ONE MINOR TIC PER TWO UNITS IN Y,
100                                      ! INTERSECT AT -10,-10,
110                                      ! ONE MAJOR TIC PER 5 MINOR TICS IN X,
120                                      ! ONE MAJOR TIC PER 2 MINOR TICS IN Y,
130                                      ! MAJOR TIC SIZE IS 5 GDU'S.
140  !
150  FRAME                               ! FRAME PLOTTING AREA.
160  END
```



GRID Statement

The GRID statement can be used as an alternative to the AXES statement when a full grid is desired.

Syntax:

GRID [$X_{tic\ spacing}$, $Y_{tic\ spacing}$ [, $X_{intersection}$, $Y_{intersection}$ [, $X_{major\ count}$, $Y_{major\ count}$ [, minor-tic size]]]]

The parameters are the same as for the AXES statement, except that the last parameter specifies the length of the minor tics only. The major tic-marks are drawn clear across the soft clipping area and a cross tic is drawn at the intersection of the minor tics. The length of the cross of the minor tics is specified by the minor tic size parameter.

Also, like the AXES statement, the grid lines and tic marks are drawn using the current line type and pen number (refer to Chapter 3).

Example

```
10   ! THIS PROGRAM DRAWS A GRID AND FRAMES THE PLOTTING AREA.
20   !
30   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY THE CRT AS PLOTTER.
40   GRAPHICS                            ! SET THE CRT GRAPHICS MODE.
50   SCALE -8,8,-8,8                     ! SPECIFY USER UNITS.
60   !
70   GRID 1,1,0,0,4,4,1                  ! DRAW GRID, SPECIFYING:
80                                       ! ONE MINOR TIC PER UNIT IN X,
90                                       ! ONE MINOR TIC PER UNIT IN Y,
100                                      ! INTERSECT AT 0,0,
110                                      ! ONE MAJOR TIC PER 4 MINOR TICS IN X,
120                                      ! ONE MAJOR TIC PER 4 MINOR TICS IN Y,
130                                      ! MINOR TIC SIZE IS 1 GDU'S.
140  !
150  FRAME                               ! FRAME PLOTTING AREA.
160  END
```


One Unit of Y
0,0 (For This Example)
One Unit of X

FRAME Statement

The FRAME statement draws a box around the current clipping area.

Syntax:

```
FRAME
```

The box is drawn using the current pen and line type around the current clipping area. The pen is positioned at the lower left corner of the frame after the operation is complete.

The second example shows an interesting use of the FRAME and CLIP statements.

Examples

```
10   ! THIS PROGRAM DRAWS A FRAME AROUND THE PLOTTING AREA.
20   !
30   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY THE CRT AS PLOTTER.
40   GRAPHICS                            ! SET THE CRT GRAPHICS MODE.
•50  FRAME                               ! FRAME PLOTTING AREA.
60   END
```



```
10   ! THIS PROGRAM USES THE CLIP AND FRAME STATEMENT TO DRAW A BAR-CHART.
20   !
30   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY THE CRT AS PLOTTER.
40   GRAPHICS                              ! SET THE CRT GRAPHICS MODE.
50   SCALE 0,21,0,100                      ! SPECIFY USER UNITS.
60   FRAME                                 ! FRAME THE PLOTTING AREA.
70   !
80   ! ***** DRAW 10 BARS AT RANDOM LENGTH.
90   FOR Bar=1 TO 10                       ! SPECIFY ONE OF TEN BARS.
100  Bar_length=RND*100                    ! RANDOM LENGTH FOR THIS EXAMPLE.
•110 CLIP Bar*2-1,Bar*2,0,Bar_length       ! CLIP PLOTTING AREA FOR FRAME.
•120 FRAME                                 ! FRAME THE BAR.
130  NEXT Bar                              ! DO IT AGAIN FOR THE NEXT BAR.
140  UNCLIP                                ! RESTORE CLIP TO SCALE SIZE.
150  END
```



LABEL and LABEL USING Statements

The LABEL and LABEL USING statements are very similar to the PRINT and PRINT USING statements except that the output list is directed to the current plotter as a label.

Syntax:

LABEL list
or
LABEL USING image specifier ; list

LABEL and LABEL USING follow the same format rules as PRINT and PRINT USING, with one exception. The exception is that an ASCII EX (CHR$(3)) terminates the labeled output at that point in the string.

For example, the program

```
10 PLOTTER IS "GRAPHICS"
20 GRAPHICS
30 MOVE 0,50
40 LABEL "THIS IS A" & CHR$(3) & "TEST"
50 END
```

results in just

```
THIS IS A
```

being labeled.

Note: The length of a short (<20 characters) unformatted label defaults to twenty characters. This is the same as the default PRINT field used by the computer. Labels will probably not be positioned as planned if they are not first formatted. Refer to PRINT and PRINT USING in the System 45 Operating and Programming Manual for formatting information.

The position and rotation of a label is controlled by the pen position and the LORG (label origin) and LDIR (label direction) statements. The size and aspect ratio of the characters in the label are controlled by the CSIZE (character size) statement.



Examples

```
10   ! THIS PROGRAM FRAMES AND LABELS THE PLOTTING AREA.
20   !
30   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY THE CRT AS PLOTTER.
40   GRAPHICS                              ! SET THE CRT GRAPHICS MODE.
50   SCALE -1,1,-1,1                       ! SPECIFY USER UNITS.
60   FRAME                                 ! FRAME PLOTTING AREA.
70   CSIZE 7                               ! SPECIFY CHARACTER SIZE.
80   LORG 5                                ! SPECIFY TO CENTER LABELS.
90   MOVE 0,0                              ! MOVE TO CENTER OF PLOTTING AREA.
100  LABEL USING "K";"Hewlett-Packard"     ! DRAW FIRST LABEL.
110  LABEL USING "K";"System 45 Desktop Computer"  ! DRAW SECOND LABEL.
120  END
```

```
10   ! THIS PROGRAM DRAWS AND LABELS THE AXES AND THEN PLOTS THE SIN OF X.
20   ! Lxaxes AND Lyaxes ARE GENERAL PURPOSE ROUTINES AND WITH A LITTLE
30   ! MODIFICATION CAN BE USED IN OTHER PROGRAMS.
40   !
50   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY THE  CRT AS PLOTTER.
60   GRAPHICS                              ! SETS CRT GRAPHICS MODE.
70   FRAME                                 ! FRAME PLOTTING AREA.
80   DEG                                   ! SET DEGREES MODE TO COMPUTE SIN(X).
90   LOCATE 15,120,15,95                   ! SPECIFY GRAPH BOUNDARIES, SPACE
100                                        ! AT LEFT AND THE BOTTOM FOR LABELS.
110                                        ! FOR LABELS (15 GDU'S).
120  !
130  ! ***** SPECIFY MAX, MIN AND LABEL STEP VALUES FOR USER UNITS.
140  Xmin=-720
150  Xmax=720
160  Xstep=180                             ! STEP SIZE FOR LABELS AND TIC MARKS.
170  !
180  Ymin=-1.5
190  Ymax=1.5
200  Ystep=.5                              ! STEP SIZE FOR LABEL AND TICS MARKS.
210  !
220  ! ***** SPECIFY USER UNITS, DRAW AXES AND FRAME PLOTTING AREA.
230  SCALE Xmin,Xmax,Ymin,Ymax             ! SPECIFY USER UNITS.
240  AXES Xstep,Ystep,0,0,10,10            ! DRAW AXES.
250  FRAME                                 ! FRAME THE PLOTTING AREA.
260  !
270  ! ***** LABEL AXES, USING SUBROUTINES.
280  GOSUB Lxaxes                          ! LABEL X AXES.
290  GOSUB Lyaxes                          ! LABEL Y AXES.
300  !
310  ! ***** PLOT SIN(X) FOR THIS EXAMPLE FROM Xmin TO Xmax.
320  FOR Xpoint=Xmin TO Xmax STEP (Xmax-Xmin)/100
330  Ypoint=SIN(Xpoint)                    ! COMPUTE SIN(X).
340  DRAW Xpoint,Ypoint                    ! DRAW A LINE TO NEXT POINT.
350  NEXT Xpoint                           ! DO IT AGAIN.
360  END                                   ! END OF EXAMPLE.
370  !
380  !
390  ! ***** SUBROUTINES TO LABEL AXES.
400  !
410 Lxaxes: !
420  CSIZE 3                               ! SPECIFY CHARACTER SIZE.
430  LDIR -90                              ! SPECIFY LABEL DIRECTION, DOWN.
440  LORG 2                                ! SPECIFY LABEL ORIGIN.
450  FOR Xposition=Xmin TO Xmax STEP Xstep
460  MOVE Xposition,Ymin                   ! MOVE TO LABEL POSITION.
470  LABEL USING "M4D.DD";Xposition        ! DRAW LABEL.
480  NEXT Xposition                        ! DO IT AGAIN, UNTIL COMPLETE.
490  RETURN
500  !
510  !
520 Lyaxes: !
530  CSIZE 3                               ! SPECIFY CHARACTER SIZE.
540  LDIR 0                                ! SPECIFY LABEL DIRECTION, L TO R.
550  LORG 8                                ! SPECIFY LABEL ORIGIN.
560  FOR Yposition=Ymin TO Ymax STEP Ystep
570  MOVE Xmin,Yposition                   ! MOVE TO LABEL POSITION.
580  LABEL USING "MD.DDX";Yposition        ! DRAW LABEL.
590  NEXT Yposition                        ! DO IT AGAIN, UNTIL COMPLETE.
600  RETURN
```


1.50
1.00
.50
0.00
-.50
-1.00
-1.50
-720.00
-540.00
-360.00
-180.00
0.00
180.00
360.00
540.00
720.00

LETTER Statement

The LETTER statement allows you to draw all keyboard alphanumerics and symbols, by typing them in on the keyboard. The size and direction of the characters are controlled the same as with the LABEL statement (also, refer to CSIZE and LDIR).

Syntax:

LETTER

The normal lower-left corner of the character is placed at the cursor position (which is not necessarily the pen position refer to CURSOR). You can preset the position of the cursor with the POINTER statement before executing the LETTER statement.

On the CRT, the cursor for the LETTER statement is a small flashing underline. The actual cursor position is at the center of flashing underline.

To move the CRT cursor press the shift or unshift of the ⇧⇧⇧, or ⇧ keys as needed. The shift of the keys moves the cursor one dot row each time the key is pressed; the unshift of these keys moves the cursor one character cell (refer to the CSIZE statement). Press HOME to move the cursor to the lower left corner.

To exit the LETTER mode, press either STEP, CONT, any special function key (such as k0) or STOP.

Example

```
10   ! THIS PROGRAM ALLOWS YOU TO DRAW CHARACTERS ON THE PLOTTING AREA
20   ! BY TYPING THEM IN ON THE KEYBOARD. THE NORMAL CURSOR CONTROLS
30   ! ARE USED TO MOVE THE CURSOR TO THE DESIRED LETTER POSITION.
40   !
50   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY THE  CRT AS PLOTTER.
60   GRAPHICS                            ! SETS CRT GRAPHICS MODE.
70   DEG                                 ! SET DEGREE MODE FOR LETTER DIRECTION.
80   LDIR 0                              ! LETTER L TO R (THIS IS THE DEFAULT).
90   CSIZE 5,.4                          ! SET CHARACTER SIZE AND ASPECT RATIO.
100  LETTER                              ! SET THE LETTER MODE.
110  END
```

LORG Statement

The LORG (label origin) statement sets the label origin position which determines where subsequent labels will be placed relative to the current pen location.

Syntax:

LORG origin position

The origin position parameter can have a value between one and nine. If LORG is not executed, the default origin position is assumed to be one.

In the following example, each number shows the initial position of the pen relative to the label when LORG has been executed with that number as its parameter. The pen positions are shown as if the label's position were fixed, though in reality the pen position is the fixed reference point.

For example LORG 7, 8, or 9 would give one of three forms of right justification.

If the label is rotated by the label direction statement (see LDIR), the relationship stays fixed relative to the label, not the plotting area.

Examples

```
 10   ! THIS PROGRAM DRAWS THE WORD "TESTS" AND THEN DRAWS SMALL NUMBERS
 20   ! AROUND THE WORD TO SHOW THE LOCATION OF THE PEN FOR THE NINE DIFFERENT
 30   ! LORG PARAMETERS.
 40   !
 50   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY CRT AS PLOTTER.
 60   GRAPHICS                              ! SET CRT GRAPHICS MODE.
 70   SCALE -2,2,-7,7                       ! SPECIFY USER UNITS.
 80   FRAME                                 ! FRAME PLOTTING AREA.
 90   CSIZE 20                              ! SPECIFY CHARACTER SIZE.
•100  LORG 5                                ! SPECIFY TO CENTER LABELS
 110  MOVE 0,0                              ! MOVE TO CENTER OF PLOTTING AREA.
•120  LABEL USING "K";"TESTS"               ! LABEL WITH THE WORD "TESTS".
 130  !
 140  ! ***** MOVE AND DRAW SMALL NUMBERS AROUND THE WORD "TESTS".
 150  CSIZE 4                               ! SPECIFY CHAR SIZE FOR NUMBERS.
 160  FOR Position=1 TO 9                   ! SPECIFY ONE OF NINE POSITIONS.
 170  MOVE INT((Position-1)/3)-1,(Position-1) MOD 3-1   ! MOVE TO POSITION.
•180  LABEL USING "D";Position              ! LABEL POSITION.
 190  NEXT Position                         ! DO IT AGAIN FOR NEXT POSITION.
 200  END
```


3
2
1
TESTS
6
5
4
9
8
7

```
10   ! THIS PROGRAM DRAWS THE WORD "TEST" USING EACH OF THE LORG PARAMETERS
20   ! AN "X" IS DRAWN TO INDICATE THE POSITION OF THE PEN FOR EACH LORG.
30   !
40   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY CRT AS PLOTTER.
50   GRAPHICS                            ! SET CRT GRAPHICS MODE.
60   SCALE 0,10,-10,0                    ! SPECIFY USER UNITS.
70   FRAME                               ! FRAME PLOTTING AREA.
80   !
90   ! ***** DRAW THE WORD "TEST" USING EACH LORG, "X" INDICATES PEN POSITION.
100  FOR I=1 TO 9                        ! SPECIFY ONE OF NINE LABEL ORIGINS.
•110 LORG 2                              ! SPECIFY LORG FOR LEFT LEBELS.
120  CSIZE 4                             ! SPECIFY CHARACTER SIZE.
130  MOVE 1,-I                           ! MOVE TO LEFT LABEL POSITION.
140  LABEL USING "5A,D,2A";"LORG ",I," =" ! LABEL LINE.
150  !
160  ! ***** MOVE TO POINT, DRAW AN "X" TO INDICATE INITIAL PEN POSITION.
•170 LORG 5                              ! SPECIFY LORG TO DRAW AN "X".
180  CSIZE 3                             ! SPECIFY CHARACTER SIZE FOR THE "X".
190  MOVE 8,-I                           ! MOVE TO POISTION FOR "X".
200  LABEL USING "K";"X"                 ! DRAW THE "X" TO INDICATE PEN POSITION
210  !
220  ! ***** MOVE TO POINT, LABEL WITH THE WORD "TEST".
•230 LORG I                              ! SPECIFY LORG FOR THE WORD "TEST".
240  CSIZE 8                             ! SPECIFY CHAR SIZE FOR THE WORD "TEST"
250  MOVE 8,-I                           ! MOVE TO POISTION OF THE "X".
260  LABEL USING "K";"TEST"              ! DRAW THE LABEL "TEST".
270  NEXT I                              ! DO IT AGAIN FOR THE NEXT LORG.
280  END
```



```
 10   ! THIS PROGRAM DEMONSTRATES THE USE OF THE LORG AND THE LDIR STATEMENTS.
 20   !
 30   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY CRT AS PLOTTER.
 40   GRAPHICS                            ! SET CRT TO GRAPHICS MODE.
 50   SCALE -1,1,-1,1                     ! SPECIFY USER UNITS.
 60   DEG                                 ! SET DEGREES MODE.
 70   FRAME                               ! FRAME THE PLOTTING AREA.
 80   !
 90   ! ***** MOVE TO POINT, DRAW AN "X" TO INDICATE PEN POSTION.
•100  LORG 5                             ! SPECIFY TO CENTER LABELS.
 110  CSIZE 3                             ! SPECIFY CHARACTER SIZE.
 120  MOVE 0,0                            ! MOVE TO CENTER OF PLOTTING AREA.
 130  LABEL USING "K";"X"                 ! DRAW AN "X" TO MARK PEN POSITION.
 140  !
 150  ! ***** MOVE TO POINT AND DRAW LABEL.
 160  MOVE 0,0                            ! MOVE TO CENTER OF PLOTTING AREA.
•170  LDIR 30                            ! SPECIFY LABEL DIRECTION AT 30 DEG.
•180  LORG 4                             ! SPECIFY LABEL ORIGIN AT BOTTOM CENTER
 190  CSIZE 10                            ! SPECIFY CHARACTER SIZE.
 200  LABEL USING "K";"TEST"              ! DRAW THE WORD "TEST".
 210  LABEL USING "K";"LABEL"             ! DRAW THE WORD "LABEL".
 220  END
```



CSIZE Statement

The CSIZE (character size) statement specifies the size and aspect ratio (character cell) of the alphanumerics or symbols to be drawn for labels.

Syntax:

CSIZE height [, aspect ratio]

The height parameter is in GDU's and must be given if the aspect ratio parameter is given. The default value is approximately 3.3 GDU's (15/4.54), which is set by the PLOTTER IS or the LIMIT statement.

The aspect ratio parameter specifies the character cell aspect ratio, defined as width/height. The default value is 9/15.



Character Cell

The width of a character cell is defined as the height times the aspect ratio.

Examples

```
10   ! THIS PROGRAM DEMONSTRATES SOME OF THE DIFFERENT CHARACTER SIZES
20   ! AND ASPECT RATIOS.
30   !
40   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY CRT AS PLOTTER.
50   GRAPHICS                              ! SET CRT GRAPHICS MODE.
60   FRAME                                 ! FRAME PLOTTING AREA.
70   SCALE -.5,10,-10,0                    ! SPECIFY USER UNITS.
80   !
90   ! ***** LABEL PLOTTING AREA WITH TWO COLUMNS OF EXAMPLE CHARACTER SIZES.
100  FOR Column=1 TO 2                     ! SPECIFY ONE OF TWO COLUMNS.
110  FOR Size=2 TO 8                       ! SPECIFY ONE OF EIGHT CHAR SIZE.
•120 CSIZE Size,Column/2                   ! SPECIFY CHAR SIZE AND ASPECT RATIO.
130  MOVE (Column-1)*4,-Size               ! MOVE TO LABEL POSITION.
•140 LABEL USING "K";"CSIZE ",Size,",",Column/2  ! LABEL PLOTTING AREA.
150  NEXT Size                             ! GET NEXT SIZE.
160  NEXT Column                           ! GET NEXT COLUMN.
170  END                                   ! END OF EXAMPLE.
```



```
10   ! THIS PROGRAM DRAWS THE COMPLETE GRAPHICS CHARACTER SET.
20   !
30   DIM A$[96]
40   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY CRT AS PLOTTER.
50   GRAPHICS                              ! SET CRT GRAPHICS MODE.
60   FRAME                                 ! FRAME THE PLOTTING AREA.
•70  CSIZE 12                              ! SPECIFY CHARACTER SIZE.
80   MOVE 5,90                             ! MOVE TO START POSITION.
90   !
100  ! ***** ASSIGN ALL PRINTABLE CHARACTERS TO A$
110  FOR C=32 TO 127
120  A$[C-31]=CHR$(C)                      ! ASSIGN CHARACTER TO A$.
130  NEXT C
140  !
150  ! ***** DRAW CHARACTERS IN GROUPS OF 16
160  FOR I=1 TO 96 STEP 16                 ! SPECIFY ONE SET OF 16 CHARACTERS.
•170 LABEL USING "K";A$[I,I+15]            ! DRAW 16 CHARACTERS.
180  NEXT I                                ! DO IT AGAIN, UNTIL COMPLETE.
190  END
```



LDIR Statement

The LDIR (label direction) statement specifies the angle at which the LETTER and the LABEL statements will draw characters.

Syntax:

LDIR angle

The angle parameter is assumed to be the counter-clockwise angle in the current angle units (DEG, RAD or GRAD) from the normal (horizontal) X axis, drawn from left to right.

The LDIR statement allows you to rotate a label about its origin (refer to LORG) by specifying the required angle.

Example

```
10   ! THIS PROGRAM DRAWS A LABEL AT EACH 30 DEGREES AROUND A CENTER POINT.
20   !
30   PLOTTER IS 13,"GRAPHICS"            ! SPECIFY CRT AS PLOTTER.
40   GRAPHICS                            ! SET CRT GRAPHICS MODE.
50   FRAME                               ! FRAME THE PLOTTING AREA.
60   DEG                                 ! SET DEGREES MODE FOR LDIR.
70   SCALE -10,10,-10,10                 ! SPECIFY USER UNITS.
80   !
90   ! ***** LABEL EACH 30 DEGREES FROM A CENTER POINT.
•100 LORG 2                              ! SPECIFY LORG, CENTER ON LEFT END.
110  FOR Degree=0 TO 330 STEP 30         ! LOOP FOR DEGREES.
•120 LDIR Degree                         ! SPECIFY A DIRECTION.
130  MOVE 0,0                            ! MOVE TO CENTER POINT.
•140 LABEL USING "K";"-------  LDIR ",Degree ! LABEL PLOTTING AREA.
150  NEXT Degree                         ! DO IT AGAIN, UNTIL COMPLETE.
160  END
```


LDIR 90
LDIR 60
LDIR 30
LDIR 0
LDIR 330
LDIR 300
LDIR 270
LDIR 240
LDIR 210
LDIR 180
LDIR 150
LDIR 120

Chapter 5
Digitizing and Cursor Control

Introduction

The statements presented in this chapter allow you to obtain information from previously plotted data. By moving the cursor around from the keyboard and then pressing (CONT) a data point can be entered into specified variables. You can preset the location and on the CRT the type of the cursor used. Another statement lets you find the coordinate of the plotters pen.

It is important to understand that the "pen" and the "cursor" are two different, separate entities. The location of each is controlled and maintained by the computer. Moving one does not affect the location of the other. On some plotting devices, (e.g., 9872A Plotter) the actual pen and cursor are physically the same, but the location maintained by the computer of each is not affected when the other is moved.

Enter the following program and then press (RUN). You will see a line drawn at random while the cursor moves in a circle. The cursor and the pen do not effect each others location.

Try inserting at line number 5 a RANDOMIZE statement to produce a different pattern.

Example

```
10   ! THIS PROGRAM HELPS DEMONSTRATES THE FACT THAT THE CURSOR AND PEN ARE
20   ! TWO SEPERATE ENTITIES, THIS IS DONE BY DISPLAYING MOVEMENT OF BOTH THE
30   ! CURSOR AND THE PEN AT THE SAME TIME.
40   !
50   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY CRT AS PLOTTER.
60   GRAPHICS                              ! SET CRT GRAPHICS MODE.
70   DEG                                   ! SET DEGREES MODE.
80   SHOW -2,2,-2,2                        ! SET USER UNITS, 0,0 IN THE CENTER.
90   FRAME                                 ! FRAME THE PLOTTING AREA.
100  MOVE 0,0                              ! MOVE PEN TO CENTER OF PLOTTER.
110  Degree=0                              ! INITILIZE Degree.
120  !
130  ! ***** DRAW RANDOM LINE AND MOVE CURSOR.
140 Loop: !
150  Direction=Direction+RND*90-45         ! RANDOMIZE NEXT DIRECTION, +/- 45 DEG.
160  PDIR Direction                        ! SET DIRECTION FOR IPLOT.
170  IPLOT .02,0                           ! IPLOT RANDOM LINE.
180  POINTER SIN(Degree),COS(Degree),1     ! REPOSITION POINTER.
190  Degree=Degree+5                       ! INCREMENT DEGREE FOR POINTER.
200  GOTO Loop                             ! DO IT AGAIN, FOREVER.
210  END
```


Cursor Location

POINTER Statement

The POINTER statement moves the cursor to a specified absolute position. On the CRT the cursor type parameter can select one of two types of cursors.

Syntax:

POINTER $X_{parameter}$, $Y_{parameter}$ [, cursor type]

The X and Y parameters are specified in current units. If the pointer is positioned outside the hard clip limit the cursor is turned off.

The cursor type parameter specifies one of two types of cursors avaiable on the CRT.

- An odd number for the cursor type parameter produces a full-screen crossed-line cursor.
- An even number for the cursor type parameter produces a small flashing cross cursor.

Once a cursor type is selected, it is the type used by all other statements that use the cursor (e.g., DIGITIZE and LIMIT without parameters) until changed again. The default cursor type for the CRT is the full-screen crossed-lines. For external plotting devices the cursor type parameter is ignored.

Examples

```
10   ! THIS PROGRAM PRODUCES A FULL-SCREEN CROSSED-LINE CURSOR (TYPE 1) AT THE
20   ! SPECIFIED POINT.
30   !
40   PLOTTER IS "GRAPHICS"              ! SPECIFY THE CRT AS PLOTTER.
50   GRAPHICS                           ! SPECIFY GRAPHICS MODE.
60   FRAME                              ! FRAME THE PLOTTING AREA.
70   SCALE 0,10,0,10                    ! SPECIFY USER UNITS.
•80  POINTER 7,6,1                      ! MOVE THE CRT CURSOR TO POINT 7,6.
90   END
```


Cursor Location

```
10   ! THIS PROGRAM PRODUCES A SMALL FLASHING-CROSS CURSOR (TYPE 0) AT THE
20   ! SPECIFIED POINT.
30   !
40   PLOTTER IS "GRAPHICS"                ! SPECIFY THE CRT AS PLOTTER.
50   GRAPHICS                             ! SPECIFY GRAPHICS MODE.
60   FRAME                                ! FRAME THE PLOTTING AREA.
70   SCALE 0,10,0,10                      ! SPECIFY USER UNITS.
•80  POINTER 4,5,0                        ! MOVE THE CRT CURSOR TO POINT 4,5.
90   END
```



CURSOR Statement

The CURSOR statement returns the cursor coordinate and pen status information to the specified variables.

Syntax:

CURSOR $X_{variable}$, $Y_{variable}$ [, pen status string variable]

The returned X and Y values are the coordinate values of the cursor in current units.

The optional third variable must be a string variable or substring, and if present it is assigned the character 0 or 1 depending on the pen's up or down status. If the pen is up, 0 is assigned to the string variable and if the pen is down, 1 is assigned to the string variable.

Remember, the pen and the cursor are NOT the same thing. On some plotters, they are physically the same (e.g., HP 9872A Plotter), but the location of the pen and cursor are kept track of separately by the computer.

Example

```
10   ! THE FOLLOWING PROGRAM LINES DEMONSTRATE HOW TO FIND AND PRINT
20   ! THE CURSOR'S LOCATION AND PEN STATUS.
30   !
40   PLOTTER IS 13,"GRAPHICS"          ! SPECIFY CRT AS PLOTTER.
50   GRAPHICS                          ! SET THE GRAPHICS MODE.
60   FRAME                             ! FRAME THE PLOTTING AREA.
70   PRINTER IS 0                      ! SPECIFY THE PRINTER.
80   !
90   POINTER 50,50                     ! SET THE CURSOR POINT FOR THIS EXAMPLE
100  !
110  !
120  CURSOR X,Y,A$                     ! INPUT CURSOR POSITION AND PEN STATUS.
130  PRINT X,Y,A$                      ! PRINT CURSOR POSITION AND PEN STATUS.
140  !
150  !
160  END
```


Cursor Location

DIGITIZE Statement

The DIGITIZE statement stops the program and allows you to reposition the cursor and then waits for you to press CONT, STEP or a special function key, at which time the coordinate values of the cursor (in current units) are assigned to the first two variables specified in the DIGITIZE statement. The third variable, if specified, is assigned pen's status information.

Syntax:

DIGITIZE $X_{variable}$, $Y_{variable}$ [, pen status string variable]

This statement is similar to the computer's INPUT statement except that a minimum of two numeric variables are assigned values from the active plotter.

The optional third variable must be a string variable or substring, and if present is assigned the character 0 or 1 depending on the pen's up or down status. If the pen is up, 0 is assigned to the string variable and if the pen is down 1 is assigned to the string variable.

On most plotting devices, the pen becomes the cursor for digitizing. On the CRT, the cursor is separate from what would be considered the pen. The location of the cursor, when it appears after executing the DIGITIZE statement, can be specified by executing a POINTER statement. The default location is the lower left corner.

Digitize From the CRT

There are two types of cursors available on the CRT: a small flashing cross and a full-screen crossed-lines. The type is selected by the third parameter of the POINTER statement. Refer to the POINTER statement.

To move the CRT cursor press the shift or unshift of the →, ↑→, or ↑ keys as needed. The shift of the keys moves the cursor one dot row each time the key is pressed, the unshift of these keys moves the cursor one character cell (refer to the CSIZE statement). Press HOME to move the cursor to the lower left corner.

To enter a digitized point from the CRT press CONT or STEP or one of the special function keys. If CONT is pressed the computer will take the coordinate of the digitized point and then continue program execution. If a special function key is pressed instead of the continue key, the computer will take the digitized point and the computer will perform the operation as defined by the special function key. Refer to the System 45 Operating and Programming Manual for more information on the operations and definitions of special function keys.

Digitize For Other Statements

If the LIMIT, LOCATE or CLIP statements are executed without parameters the computer will automatically enter a digitize mode and wait for you to enter two opposite corners for the executed statement. The cursor on the CRT can be moved and the point entered as described in the above two paragraphs. On external plotter, refer to the plotters operating manual for the digitize procedure.

Examples

```
10   ! THIS PROGRAM DEMONSTRATES THE USE OF THE DIGITIZE STATEMENT. SEVERAL
20   ! LINES ARE DRAWN ON THE CRT, YOU CAN FIND THE COORDINATES OF ANY POINT
30   ! BY POSITIONING THE CURSOR OVER THE POINT AND THEN PRESSING THE CONTINUE
40   ! KEY. NOTE: TO CHANGE CURSOR TYPE PRESS SPECIAL FUNCTION KEY #0.
50   !
60   PRINTER IS 0                          ! SPECIFY PRINTOUT DEVICE.
70   FIXED 2                               ! SPECIFY FIXED 2 FOR PRINTOUT.
80   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY CRT AS PLOTTER.
90   GRAPHICS                              ! SET CRT GRAPHICS MODE.
100  FRAME                                 ! FRAME PLOTTING AREA.
110  !
120  ON KEY #0 GOSUB Change_cursor         ! PRESS KEY #0 TO CHANGE CURSOR TYPE.
130  Cursor_type=0                         ! INITIALIZE CURSOR_TYPE VARIABLE.
140  !
150  ! ***** DRAW A LINE TO GDU 100,100 FOR THIS EXAMPLE.
160  DRAW 100,100                          ! DRAW LINE.
170  !
180  ! ***** DRAW A TRIANGLE FOR THIS EXAMPLE.
190  MOVE 60,50
200  DRAW 80,10
210  DRAW 40,10
220  DRAW 60,50
230  !
240  ! ***** INITIALIZE CURSOR POSITION AND TYPE.
250  POINTER 50,40,Cursor_type             ! POSITION CURSOR.
260  !
270  !
280  ! ***** LOOP TO DIGITIZE POINTS ON THE CRT.
290 Loop: !
•300 DIGITIZE X,Y,A$                       ! INPUT CURSOR COORDINATE & PEN STATUS.
310  IF Cursor_type<>Old_cursor_type THEN 330  ! SKIP PRINTOUT IF CURSOR CHANGE.
320  PRINT X,Y,A$                          ! PRINT COORDINATE.
330  Old_cursor_type=Cursor_type           ! ASSGIN VALUE TO OLD_CURSOR_TYPE.
340  GOTO Loop                             ! DO IT AGAIN, FOREVER.
350  End
360  !
370  ! ***** CHANGE CURSOR TYPE WHEN KEY # 0 IS PRESSED.
380 Change_cursor: !
390  Cursor_type=NOT Cursor_type           ! CHANGE CURSOR_TYPE VARIABLE.
400  CURSOR Xc,Yc                          ! FIND CURRENT CURSOR COORDINATE.
410  POINTER Xc,Yc,Cursor_type             ! REPLACE CURSOR WITH NEW TYPE.
420  RETURN                                ! RETURN TO LOOP.
```




```
10   ! THIS PROGRAM TRANSFERS DIGITIZED POINTS FROM 9872A TO CRT.
20   ! MOVE THE CURSOR (PEN) WITH THE PEN MOVEMENT KEY ON THE 9872A PLOTTER
30   ! AND THEN PRESS 'ENTER' ON THE 9872A PLOTTER TO ENTER POINT.
40   !
50   DIM Pen$[10]
60   PRINTER IS 0                        ! SPECIFY INTERNAL PRINTER TO PRINT DATA.
70   !
80   PLOTTER IS 7,5,"9872A"              ! SPECIFY 9872A AS PLOTTING DEVICE.
90   DISP "DIGITIZE LIMITS, PRESS 'ENTER' ON 9872A AFTER EACH POINT."
100  LIMIT                               ! DIGITIZE LIMITS ON 9872A.
110  FRAME                               ! FRAME LIMITS ON 9872A.
120  SHOW -1,1,-1,1                      ! DEFINE UDU'S FOR 9872A.
130  !
140  PLOTTER IS 13,"GRAPHICS"            ! SPECIFY CRT AS PLOTTING DEVICE.
```

```
 150  GRAPHICS                          ! SPECIFY GRAPHICS MODE
 160  SHOW -1,1,-1,1                    ! DEFINE UDU'S FOR CRT.
 170  FRAME                             ! FRAME LIMITS ON CRT.
 180  !
 190 Loop: !
 200  PLOTTER 7,5 IS ON                 ! ENABLE 9872A.
 210  Pen$=""                           ! CLEAR Pen$.
•220  DIGITIZE X,Y,Pen$                 ! DIGITIZE POINT, AND GET PEN STATUS.
 230  PRINT X,Y,Pen$                    ! PRINT X,Y AND PEN STATUS.
 240  !
 250  PLOTTER 13 IS ON                  ! ENABLE CRT.
 260  PLOT X,Y                          ! PLOT POINT.
 270  !
 280  GOTO Loop                         ! DO IT AGAIN, FOREVER.
 290  END
```



WHERE Statement

The WHERE statement returns the coordinate values in current units of the last plotted or moved to point. If the last point is outside the clip boundaries the returned value will not reflect the actual pen position but rather the last plotted or moved to point.

Syntax:

WHERE $X_{variable}$, $Y_{variable}$ [, pen status string variable]

The returned X and Y values are the coordinates of the pen position in current units. If the pen was sent outside of the clipping area, that coordinate would be returned.

The optional third variable must be a string variable or a substring, and if present is assigned the character 0 or 1 depending on the pen status. If the pen is up, 0 is assigned to the string variable; and if the pen is down, 1 is assigned to the string variable.

Example

```
10   !
20   !
•30  WHERE X,Y,A$                          ! RETURN THE COORDINATE OF THE
40                                         ! LAST PLOTTED POINT AND PEN STATUS.
50   !
60   PRINT X,Y,A$                          ! PRINT X,Y AND PEN STATUS.
70   !
80   !
90   ! DO NOT RUN THIS PROGRAM, THESE ARE ONLY EXAMPLE PROGRAM LINES.
```

Chapter 6
CRT Graphics Control Statements

Introduction

The statements described in this chapter are used only in conjunction with the CRT and not with other plotters or displays. Two of the statements allow you to select one of two CRT modes (graphics or alpha). Another statement allows you to copy the graphic display onto the computer's internal printer for hard copy output. Two other statements allow you to copy the graphics display to-or-from the computer's main memory as an integer array, where you can store it for display later.

Once in an integer array, the information can be stored on the tape cartridge and then later redisplayed by executing just a few statements.

GRAPHICS and EXIT GRAPHICS Statements

The GRAPHICS and EXIT GRAPHICS statements turn the CRT graphics mode on and off respectively. If you are plotting on the CRT and want to see what you are plotting, GRAPHICS is the one statement that has to be executed. You can plot on the CRT while it is in the alpha mode and then switch to graphics mode to see what you have plotted.

Syntax:

```
GRAPHICS
```

or

```
EXIT GRAPHICS
```

The CRT remains in the graphic mode until EXIT GRAPHICS is executed, an error occurs, a typing key or the recall key is pressed on the keyboard.

Example

```
10   !
20   !
30   GRAPHICS                                  ! SET CRT GRAPHICS MODE.
40   !
50   !
60   !
70   EXIT GRAPHICS                             ! SET CRT ALPHANUMERIC MODE.
80   !
90   !
100  ! DO NOT RUN THIS PROGRAM, THESE ARE ONLY EXAMPLE PROGRAM LINES.
```

DUMP GRAPHICS Statement

The DUMP GRAPHICS statement allows a graphic display on the CRT to be copied onto the internal printer.

Syntax:

DUMP GRAPHICS [lower bound, [upper bound]]

This statement can be executed by the program or from the keyboard while the CRT is in or out of the graphics mode.

The two optional parameters allows you to copy any horizontal area displayed on the CRT by specifying in current units the upper and lower bounds of the area to be copied.

Without the second parameter, the bottom of the CRT will be used as the lower bound.

Without the first and second parameters, the entire CRT display is copied onto the internal printer.

Example

```
 10   !
 20   !
•30   DUMP GRAPHICS                    ! TRANSFER ALL OF THE CRT GRAPHICS
 40                                     ! DISPLAY TO THE INTERNAL PRINTER.
 50   !
 60   !
 70   !
 80   !
•90   DUMP GRAPHICS -2,2               ! DUMP FROM -2 UNITS TO +2 UNITS OF
 100                                    ! THE CRT GRAPHICS DISPLAY TO THE
 110                                    ! INTERNAL PRINTER.
 120  !
 130  !
 140  !
 150  ! DO NOT RUN THIS PROGRAM, THESE ARE ONLY EXAMPLE PROGRAM LINES.
```

GLOAD and GSTORE Statements

The GLOAD (Graphic Load) and GSTORE (Graphics Store) statements are made available by the Mass Memory ROM, but are described here because they are used with graphic. The Mass Storage ROM must be installed before these statements can be used. Refer to the Mass Storage Techniques Manual (P/N 09845-90070).

The GLOAD and GSTORE statements allow you to copy the CRT memory to-or-from an integer array within the computer's memory. The CRT memory contains the data that produces dots, which make up the CRT graphics display. When copied into the computer's memory the data can be stored (GSTORE) and then later copied back (GLOAD) into the CRT memory for display.

A complex graphics display may take several minutes to generate using the plotting statements, but by using GSTORE and then later GLOAD, the same display can be re-displayed in a fraction of a second.

Syntax:

GLOAD integer array name (*)
or
GSTORE integer array name (*)

The integer array used with these statements must have 36 elements for each horizontal dot row, plus one extra element for a pointer. The pointer is the first element of the array.

If you want to store the entire graphics display, you need to dimension an integer array with 16381 elements. One element is for the pointer, plus 36 elements for each row, and there are 455 rows (1+36x455=16381). If you are using OPTION BASE 0 (refer to the System 45 Operating and Programming Manual) the subscript of the first element of the array will be zero (e.g., A(0)) and the dimension size could be as shown.

```
10 OPTION BASE 0
20 INTEGER A(16380)
```

Zero through 16380 equals 16381 elements.

If it is not necessary to copy the entire graphics display you can dimension the integer array smaller. An integer array with 8173 elements (1+227x36=8173) would hold about one half of the graphics display (227 dot rows).

The first element (i.e., A(0)) will always be used as the load or store pointer and must be a negative number. The value that you assign to the pointer determines where the load or store operation will begin.

The meaningful range of value of the pointer is −1 to −16379 as shown below.





The number represents the starting address that must be assigned to the pointer to start a load or store operation at the location represented by the square.

The following expression can be used to compute the starting address for a given dot row. Note: the assigned value must be negative.

A(0)=−(Rx36+1)

where the variable R is the row number (0 to 454) that you want to start the load or store operation. The top row number is zero (0) and the bottom row number is 454.

Enter the following program and then press (RUN).

```
10   ! THIS PROGRAM DRAWS A PATTERN ACROSS THE TOP OF THE CRT USING THE NORMAL
20   ! PLOTTING STATEMENTS. THE PATTERN IS THEN REPEATED DOWN THE SCREEN USING
30   ! THE GSTORE AND GLOAD STATEMENTS. THE PATTERN IS REPEATED MUCH FASTER
40   ! THEN IT WAS DRAWN THE FIRST TIME.
50   !
60   OPTION BASE 0                          ! SPECIFY BASE ZERO (DEFAULT).
70   INTEGER A(900)                         ! SPECIFY INT. ARRAY TO STORE GRAPHICS.
80                                          ! 25*36 ELEMENTS PLUS THE A(0) ELEMENT.
90   PLOTTER IS 13,"GRAPHICS"               ! SPECIFY THE CRT AS PLOTTER.
100  GRAPHICS                               ! SET GRAPHICS MODE.
110  DEG                                    ! SET DEGREES MODE.
120  SCALE 0,25*360,-10,10                  ! SPECIFY USER UNITS.
130  !
140  ! ***** DRAW PATTERN FOR THIS EXAMPLE.
150  MOVE 0,9.5                             ! MOVE TO START.
160  FOR X=0 TO 25*360 STEP 90              ! FOR..NEXT TO DRAW PATTERN.
170  DRAW X,.5*SIN(X)+9.5                   ! DRAW LINE.
180  NEXT X                                 ! DO IT AGAIN UNTIL COMPLETE.
190  !
200  ! ***** STORE PATTERN IN AN INTEGER ARRAY.
•210 A(0)=-1                                ! SPECIFY GSTORE START POINT.
•220 GSTORE A(*)                            ! GSTORE PATTERN IN A(*) ARRAY.
230  !
240  ! ***** REPEAT PATTERN DOWN THE CRT 15 TIMES USING GLOAD.
245  Width=25                               ! SPECIFY THE WIDTH OF THE PATTERN.
250  FOR Pattern=1 TO 15
270  A(0)=-(Pattern*Width*36+1)             ! SPECIFY GLOAD START POSITION.
•280 GLOAD A(*)                             ! GLOAD THE PATTERN ONTO THE CRT.
290  NEXT Pattern                           ! DO IT AGAIN UNTIL COMPLETE.
300  END
```

The program will produce a pattern across the top of the CRT at the normal line generating speed. And then by using GSTORE and GLOAD, the pattern will be repeated 15 times down the screen (25 dot rows at a time). As you will be able to see, the pattern can be repeated faster than it can be generated the first time.



Appendix A
Reflected Plots

Reflected plots are produced by exchanging one or more sets of parameters in the following graphics statement.

```
LIMIT
SHOW
SCALE
```

With each of those statements, the parameters are in the order of

X1,X2,Y1,Y2

If you enter the parameters as X2,X1,Y1,Y2 the resulting plot is reflected across the Y-axis.

If you enter the parameters as X1,X2,Y2,Y1 the resulting plot is reflected across the X-axis.

If you enter the parameters as X2,X1,Y2,Y1 the resulting plot is reflected through the origin.

If you are labeling a reflected plot, be careful:

the LIMIT statement reflects the labeled characters as well as the plot.

the SHOW and SCALE statements position the labels relative to the X,Y position called for by a MOVE regardless of how that looks due to the reflection.

The following illustrations show the three types of reflected plots:



Appendix B
External Plotter Installation

Introduction

Hewlett-Packard and several other manufacturers have external plotters that can be connected to the System 45. The following information will aid in connecting the HP 9872A Plotter and some of the other manufacturers' incremental plotters to the computer.

9872A Plotter Installation

The 9872A Plotter can be used as an external plotter with the System 45. Refer to the PLOTTER IS statement for information on how to select the 9872A as the active plotter.

Ensure that the interface select code switch and plotter address code switches are set to the proper positions as indicated below.

Interface Select Code

The 9872A Plotter is connected to the System 45 via an HP 98034A HP-IB Interface. Refer to the HP 98034A HP-IB Interface Installation and Service Manual for complete information about the interface and cables.

The interface is preset to a select code of 7 at the factory. To change the setting, rotate the switch (shown below) using a small screwdriver.



Select Code Switch

Plotter Address Code

Since each HP-IB interface can have as many as 14 devices connected to it, each device must be set to a specific address code.

The plotter can be set to any one of 31 HP-IB addresses ranging from 0 thru 30. Each address can be selected by setting the switches on the plotter back panel (shown next) to the appropriate binary bit positions for the particular address value desired.

The plotter is set to an address code of 05 at the factory. Check your plotter for the proper switch positions shown below.



Plotter Address Switches

The following table lists the switch positions for each address value.

Address Switch Positions

Address Characters		Address Switch Settings					Address Codes	
Listen	Talk	(5)	(4)	(3)	(2)	(1)	decimal	octal
SP	@	0	0	0	0	0	0	0
!	A	0	0	0	0	1	1	1
"	B	0	0	0	1	0	2	2
#	C	0	0	0	1	1	3	3
$	D	0	0	1	0	0	4	4
%	E	0	0	1	0	1	5	5 ← preset
&	F	0	0	1	1	0	6	6
'	G	0	0	1	1	1	7	7
(	H	0	1	0	0	0	8	10
)	I	0	1	0	0	1	9	11
*	J	0	1	0	1	0	10	12
+	K	0	1	0	1	1	11	13
,	L	0	1	1	0	0	12	14
−	M	0	1	1	0	1	13	15
.	N	0	1	1	1	0	14	16
/	O	0	1	1	1	1	15	17
0	P	1	0	0	0	0	16	20
1	Q	1	0	0	0	1	17	21
2	R	1	0	0	1	0	18	22
3	S	1	0	0	1	1	19	23
4	T	1	0	1	0	0	20	24
5	U	1	0	1	0	1	21	25
6	V	1	0	1	1	0	22	26
7	W	1	0	1	1	1	23	27
8	X	1	1	0	0	0	24	30
9	Y	1	1	0	0	1	25	31
:	Z	1	1	0	1	0	26	32
;	[	1	1	0	1	1	27	33
<	/	1	1	1	0	0	28	34
=	]	1	1	1	0	1	29	35
>	^	1	1	1	1	0	30	36

Incremental Plotter Installation

An incremental plotter can be used as an external plotter with the System 45, using the 98040A Interface. Refer to the PLOTTER IS statement for information on how to select the incremental plotter as the plotter.

Ensure that the interface select code switch is set to the proper position.

Refer to the 98040A Interface Installation and Service Manual (P/N 98040-90000) for information concerning the select code switch and a list of incremental plotters that can be used with this interface.

Appendix C
Other Example Programs

Introduction

The programs in this section are given to show some interesting capabilities and programming techniques. In most cases, each program line has a comment that describes what the line does. After running the program and understanding how it works, change a few lines to make the program do something different or to fit one of your applications.

```
10   ! THIS PROGRAM DRAWS A FOOTBALL FIELD TO DEMONSTRATE SOME INTERESTING
20   ! CAPABILITIES AND PROGRAMMING TECHNIQUES USING THE GRAPHIC STATEMENTS.
30   !
40   PLOTTER IS 13,"GRAPHICS"          ! SPECIFY THE CRT AS PLOTTER.
50   GRAPHICS                          ! SET GRAPHICS MODE.
60   FRAME                             ! FRAME PLOTTING AREA.
70   LOCATE 10,112,10,90               ! SPECIFY NEW PLOTTING AREA.
80   SHOW -60,60,-80/3,80/3            ! SPECIFY OVERALL SIZE.
90   DEG                               ! SET DEGREES MODE.
100  !
110  ! ***** DRAW YARD LINES.
120  CLIP -40,40,-80/3,80/3            ! CLIP LENGHT OF YARD LINES.
130  LINE TYPE 3                       ! SPECIFY DASH LINE.
140  GRID 10,70,-40,-30                ! DRAW YARD LINES.
150  !
160  ! ***** DRAW FIELD PERIMETER.
170  CLIP -50,50,-80/3,80/3            ! CLIP AT FIELD PERIMETER.
180  LINE TYPE 1                       ! SPECIFY SOLID LINE.
190  FRAME                             ! DRAW PERIMETER.
200  !
210  ! ***** DRAW END ZONES.
220  CLIP -60,60,-80/3,80/3            ! CLIP AT END ZONES.
230  FRAME                             ! DRAW PERIMETER AROUND END ZONES.
240  !
250  ! ***** DRAW GOALPOSTS.
260  CSIZE 3                           ! SPECIFY CHARACTER SIZE.
270  LORG 5                            ! CENTER LABELS.
280  LDIR 0                            ! LABEL LEFT TO RIGHT.
290  FOR Goal=-1 TO 1 STEP 2           ! SPECIFY ONE OF TWO GOALS.
300  FOR Post=-1 TO 1 STEP 2           ! SPECIFY ONE OF TWO POSTS.
310  MOVE Goal*60,Post*6               ! MOVE TO GOALPOST POSITION.
320  LABEL USING "K";"O"               ! DRAW GOALPOST.
330  NEXT Post                         ! DO IT AGAIN FOR OTHER POST.
340  NEXT Goal                         ! DO IT AGAIN FOR OTHER GOAL.
350  !
360  ! ***** LABEL YARD LINES.
370  CSIZE 4                           ! SPECIFY CHARACTER SIZE.
380  LORG 5                            ! CENTER LABELS.
390  FOR Side=-1 TO 1 STEP 2           ! SPECIFY ONE OF TWO FIELD SIDES.
400  FOR Yard=-40 TO 40 STEP 10        ! SPECIFY YARD LINE.
410  LDIR 90+90*SGN(Side)              ! INVERT LABEL DEPENDING ON SIDE.
420  MOVE Yard,Side*25                 ! MOVE TO YARD AND SIDE.
430  LABEL USING "DD";50-ABS(Yard)     ! LABEL YARD LINE.
440  NEXT Yard                         ! DO IT AGAIN FOR NEXT YARD LINE.
450  NEXT Side                         ! DO IT AGAIN FOR OTHER SIDE.
460  !
470  ! ***** LABEL END ZONES.
480  CSIZE 8                           ! SET CHARACTER SIZE.
490  LORG 5                            ! CENTER LABELS.
500  LDIR 90                           ! LABEL BOTTOM TO TOP.
510  MOVE -55,0                        ! MOVE TO LEFT END ZONE.
520  LABEL USING "K";"HOME"            ! LABEL END ZONE.
530  LDIR -90                          ! LABEL TOP TO BOTTOM.
540  MOVE 55,0                         ! MOVE TO RIGHT END ZONE.
550  LABEL USING "K";"VISITORS"        ! LABEL RIGHT END ZONE.
560  !
570  ! ***** LABEL SCORE
580  CSIZE 5                           ! SET CHARACTER SIZE.
590  LORG 5                            ! CENTER LABELS.
600  LDIR 0                            ! LABEL LEFT TO RIGHT.
610  MOVE 0,-35                        ! MOVE TO BOTTOM OF PLOTTING AREA.
620  LABEL USING "K";"HOME 21, VISITORS 14"   ! LABEL SCORE.
630  END
```


HOME
VISITORS
10 20 30 40 50 40 30 20 10
10 20 30 40 50 40 30 20 10
HOME 21, VISITORS 14

```
10   ! THIS PROGRAM PLOTS THE SINE(X)/X FROM -4PI TO +4PI TO DEMONSTRATE
20   ! SOME INTERESTING CAPABILITY AND PROGRAMMING TECHNIQUES.
30   ! THE PROGRAM WILL STOP AND WAIT FOR YOU TO MOVE THE CURSOR TO DIGITIZE
40   ! A POINT, AND THEN PRINT THE "X" AND "Y" VALUES.
50   !
60   PLOTTER IS 13,"GRAPHICS"              ! SPECIFY THE CRT AS PLOTTER.
70   GRAPHICS                              ! SET GRAPHICS MODE.
80   FRAME                                 ! FRAME PLOTTING AREA.
90   PRINTER IS 0                          ! SPECIFY INTERNAL PRINTER.
100  RAD                                   ! SET RADIANS MODE.
110  LOCATE 10,120,20,90                   ! SPECIFY GRAPH BOUNDARIES.
120  !
130  ! ***** SPECIFY MAXIMUM AND MINIMUM VALUES FOR USER UNITS.
140  Xmin=-4*PI
150  Xmax=4*PI
160  Ymin=-.5
170  Ymax=1.5
180  !
190  ! ***** SPECIFY USER UNITS, DRAW AXES AND FRAME GRAPH.
200  SCALE Xmin,Xmax,Ymin,Ymax             ! SPECIFY USER UNITS.
210  AXES PI/6,.5,0,0,3,10                 ! DRAW AXES.
220  FRAME                                 ! FRAME PLOTTING AREA.
230  !
240  ! ***** LABEL AXES, USING SUBROUTINES.
250  GOSUB Lxaxis                          ! LABEL X AXIS.
260  GOSUB Lyaxis                          ! LABEL Y AXIS.
270  !
280  ! ***** PLOT FUNCTION WITH FILL TO BASE LINE.
290  FOR X=Xmin TO Xmax STEP PI/20
300  IF X=Xmin THEN MOVE X,SIN(S)/X        ! MOVE TO START IF X=Xmin.
310  IF X=0 THEN 350                       ! AVOID DIVIDING BY ZERO.
320  DRAW X,SIN(X)/X                       ! DRAW TO NEXT POINT ON CURVE.
330  MOVE X,0                              ! MOVE TO BASE LINE.
340  DRAW X,SIN(X)/X                       ! DRAW BACK TO CURVE FOR FILL.
350  NEXT X                                ! NEXT POINT ON CURVE.
360  !
370  ! ***** MOVE AND LABEL PLOT
380  MOVE -3.5*PI,1.1                      ! MOVE TO START OF LABEL.
390  LORG 1                                ! SPECIFY LABEL ORIGIN.
400  CSIZE 8                               ! SPECIFY CHARACTER SIZE.
410  LABEL USING "K";"SIN (X)/X"           ! LABEL PLOT.
420  CSIZE 3                               ! RESET CHAR SIZE FOR DIGITIZE.
430  POINTER -2*PI,.5,0                    ! PUT POINTER AT 0,0 AS SMALL CROSS.
440  !
450  ! ***** THIS ALLOWS YOU TO DIGITIZE AND PRINT POINTS FROM THE CRT.
460 Loop: !
470  DIGITIZE X,Y                          ! WAIT FOR CONTINUE THEN INPUT X AND Y
480                                        ! & PRINT X AND Y ON SPECIFIED PRINTER.
490  PRINT USING "2(K,X#D.2D),2/";"X=",X/PI," PI,     Y=",Y
500  GOTO Loop                             ! DO IT AGAIN, FOREVER.
510  END
520  !
530  !
540  ! ***** LABEL AXES ROUTINES.
550 Lxaxis: !
560  CSIZE 3                               ! SPECIFY CHARACTER SIZE.
570  LDIR -(PI/2)                          ! SPECIFY LABEL DIRECTION, DOWN.
580  LORG 2                                ! SPECIFY LABEL ORIGIN.
590  FOR X1=Xmin TO Xmax STEP PI
600  MOVE X1,Ymin                          ! MOVE TO LABEL POSITION.
```

```
610  LABEL USING "M4DX,K";X1/PI,"PI"      ! DRAW LABEL.
620  NEXT X1                              ! DO IT AGAIN, UNTIL COMPLETE.
630  RETURN
640  !
650 Lyaxis: !
660  CSIZE 3                              ! SPECIFY CHARACTER SIZE.
670  LDIR 0                               ! SPECIFY LABEL DIRECTION, L TO R.
680  LORG 8                               ! SPECIFY LABEL ORIGIN.
690  FOR Y1=Ymin TO Ymax STEP .5
700  MOVE Xmin,Y1                         ! MOVE TO LABEL POSITION.
710  LABEL USING "MD.DDX";Y1              ! DRAW LABEL.
720  NEXT Y1                              ! DO IT AGAIN, UNTIL COMPLETE.
730  RETURN
```



```
10   ! THIS PROGRAM HELPS DEMONSTRATES HOW THE SHOW STATEMENT
20   ! DEFINES THE PLOTTING AREA.
30   ! IN THIS EXAMPLE A SQUARE 20 BY 20 WITH POINT 0,0 AT THE CENTER
40   ! IS SPECIFIED BY THE SHOW STATEMENT.
50   !
60   ! THE PROGRAM WILL STOP AND WAIT FOR YOU TO DIGITIZE THE LIMITS OF
70   ! THE PLOTTING AREA. NOTE HOW THE FINAL NUMBER OF USER UNITS ARE
80   ! EFFECTED BY THE ASPECT RATIO OF THE AREA THAT YOU SPECIFY.
90   !
100  PLOTTER IS 13,"GRAPHICS"            ! SPECIFIES THE CRT AS PLOTTER.
110  GRAPHICS                            ! SETS CRT GRAPHICS MODE.
120  DEG                                 ! SETS DEGREES MODE.
130  !
140  ! ***** DIGITIZE THE SIZE AND SHAPE OF THE PLOTTING AREA.
150  !       MOVE THE CURSOR AROUND AND PRESS CONTINUE TO INPUT
160  !       TWO OPPOSITE CORNERS.
170  LIMIT                               ! ALLOWS YOU TO DEFINE SIZE AND SHAPE.
180  !
190  ! ***** FRAME USER SPECIFIED PLOTTING AREA.
200  LINE TYPE 3                         ! SPECIFY DASHED LINE.
210  FRAME                               ! FRAME PLOTTING AREA.
220  LINE TYPE 1                         ! SPECIFY SOLID LINE.
230  !
240  ! ***** FIND ASPECT RATIO OF USER SPECIFIED PLOTTING AREA.
250  !       INITIALIZE VALUES FOR: Xmin, Xmax, Ymin AND Ymax.
260  Xmin=0                              ! INITIALIZE Xmin.
270  Xmax=100                            ! INITIALIZE Xmax.
280  Ymin=0                              ! INITIALIZE Ymin.
290  Ymax=100                            ! INITIALIZE Ymax.
300  IF RATIO>1 THEN Xmax=100*RATIO      ! ADJUST Xmax IF RATIO>1.
310  IF RATIO<1 THEN Ymax=100/RATIO      ! ADJUST Ymax IF RATIO<1.
320  Midx=Xmax/2                         ! FIND MID POINT OF Xmax.
330  Midy=Ymax/2                         ! FIND MID POINT OF Ymax.
340  Mids=MIN(Midx,Midy)                 ! FIND MID POINT OF SHORT SIDE.
350  !
360  ! ***** DRAW AN EXPANDING PLOT FOUR TIMES.
370  FOR Offset=Mids/4 TO Mids STEP Mids/4  ! OFFSET FOR CORNERS OF LOCATE.
380  LOCATE Midx-Offset,Midx+Offset,Midy-Offset,Midy+Offset ! SPECIFY SIZE.
390  SHOW -10,10,-10,10                  ! USER UNITS FOR EXPANDING SQUARE.
400  !
410  PEN 1                               ! SPECIFIES A PEN THAT DRAWS.
420  GOSUB Plot                          ! GOSUB TO THE PLOT SUBROUTINE TO DRAW.
430  IF Offset>=Mids THEN Wrap_up        ! DO NO ERASE, GOTO Wrap_up.
440  PEN -1                              ! SPECIFIES AN ERASER.
450  GOSUB Plot                          ! GOSUB TO PLOT SUBROUTINE TO ERASE.
460  NEXT Offset                         ! LOOP TO DO IT AGAIN.
470  !
480  ! **** COMPLETE THE EXAMPLE
490 Wrap_up: !
500  !       FRAME THE 20 X 20 SQUARE WITH A DASHED LINE.
510  LINE TYPE 3                         ! SPECIFY DASHED LINE.
520  FRAME                               ! FRAME THE SQUARE.
530  !       FRAME THE USER SPECIFIED PLOTTING AREA WITH SOLID LINE.
540  LINE TYPE 1                         ! SPECIFY SOLID LINE.
550  LOCATE Xmin,Xmax,Ymin,Ymax
560  FRAME                               ! FRAME PLOTTING AREA.
570  SHOW -10,10,-10,10                  ! ASSIGN USER UNITS.
580  !       REPLOT TO FILL IN AXES AND CIRCLES AS NEEDED.
590  GOSUB Plot                          ! DRAW IT AGAIN TO FILL IN.
600  END                                 ! EXAMPLE COMPLETE.
```

```
610  !
620  !
630  !
640  ! ***** SUBROUTINE, DRAW OR ERASE PLOT DEPENDING ON PEN.
650 Plot: FRAME                              ! FRAME EXPANDING SQUARE.
660  AXES 1,1,0,0,5,5,5                      ! DRAW AXES.
670  FOR Circle=1 TO 3                       ! DRAW OR ERASE THREE CIRCLES.
680  IF Circle=1 THEN MOVE 10,5              ! MOVE TO POSITION OF CIRCLE ONE.
690  IF Circle=2 THEN MOVE -5,10             ! MOVE TO POSITION OF CIRCLE TWO.
700  IF Circle=3 THEN MOVE -5,-5             ! MOVE TO POSITION OF CIRCLE THREE.
710  !       DRAW CIRCLE AT SPECIFIED POSITION.
720  FOR Arc=0 TO 360 STEP 15                ! SPECIFY 15 DEG PER ARC.
730  PDIR Arc+90                             ! DIRECTION TO NEXT POINT ON ARC.
740  RPLOT 3,0                               ! DRAW PART OF CIRCLE.
750  NEXT Arc                                ! NEXT ARC.
760  NEXT Circle                             ! NEXT CIRCLE.
770  RETURN                                  ! RETURN FROM SUBROUTINE.
780  END
```









Appendix D
Syntax Summary

AXES [$X_{tic\ spacing}$, $Y_{tic\ spacing}$ [, $X_{intersection}$, $Y_{intersection}$ [, $X_{major\ count}$, $Y_{major\ count}$ [, major-tic size]]]]

The AXES statement draws a pair of axes with optional (linearly spaced) tic marks.

CLIP [X_{min} , X_{max} , Y_{min} , Y_{max}]

The CLIP statement defines the soft clip boundaries. Omitting the parameters allows any two diagonal corners to be digitized.

CSIZE height [, aspect ratio]

The CSIZE (character size) statement is used to specify the size and aspect ratio of characters used in labels. The height defaults to approximately 3.3 GDU's (15/4.54). The aspect ratio (width/height) defaults to 9/15.

CURSOR $X_{variable}$, $Y_{variable}$ [, pen status string variable]

The CURSOR statement returns the values of the cursor coordinate to the specified variables. Pen status is assigned to the string variable; 0 for up, 1 for down.

DIGITIZE $X_{variable}$, $Y_{variable}$ [, pen status string variable]

The DIGITIZE statement pauses program execution and allows you to reposition the cursor; execution is resumed by pressing the CONTINUE key, any SFK or the STEP key. The coordinate of the cursor is assigned to the specified variables. Pen status is assigned to the string variable; 0 for up, 1 for down.

DRAW $X_{parameter}$, $Y_{parameter}$

The DRAW statement drops the pen and moves it to the absolute X,Y coordinate.

DUMP GRAPHICS [lower bound [, upper bound]

The DUMP GRAPHICS statement copies the CRT graphic display onto the internal thermal printer. Any horizontal area can be copied by specifying its upper and/or lower bound in current units.

EXIT GRAPHICS

The EXIT GRAPHICS statement returns the CRT to the normal alphanumerics mode from the graphics mode.

FRAME

The FRAME statement draws a box around the current clipping area.

GCLEAR [paper advance]

The GCLEAR statement clears the CRT of previously plotted data. The paper advance value specifies how many millimetres of paper to advance on certain plotters; it has no effect on CRT graphics.

GLOAD integer array name (*)

The GLOAD (graphics load) statement is made available by the Mass Memory ROM, but is described in this manual because it is used with graphics. GLOAD transfers data from an integer array to the CRT graphics memory. The first element in the integer array is the load pointer.

GRAPHICS

The GRAPHICS statement sets the CRT to the graphics mode.

GRID [$X_{\text{tic spacing}}$, $Y_{\text{tic spacing}}$ [, $X_{\text{intersection}}$, $Y_{\text{intersection}}$ [, $X_{\text{major count}}$, $Y_{\text{major count}}$ [, minor-tic size]]]]

The GRID statement can be used as an alternative to the AXES statement and is used to draw a full screen grid.

GSTORE integer array name (*)

The GSTORE (graphics store) statement is made available by the Mass Memory ROM, but is described in this manual because it is used with graphics. GSTORE transfers data from the CRT graphics memory to an integer array. The first element in the integer is the store pointer.

IPLOT $X_{increment}$, $Y_{increment}$ [, pen control]

The IPLOT statement allows incremental plotting from the last plotted point. The pen control is the same as for the PLOT statement.

LABEL list

The LABEL statement is used like the PRINT statement to draw labels on the plotter.

LABEL USING image specifier; list

The LABEL USING statement is used like the PRINT USING statement to draw formatted labels on the plotter.

LDIR angle

The LDIR statement specifies the angle at which subsequent labels will be drawn. The angle specifies counter-clockwise rotation of the label from the positive X-axis in current angular units.

LETTER

The LETTER statement allows you to draw all keyboard alphanumerics and symbols, by typing them in on the keyboard.

LIMIT [X_{min}, X_{max}, Y_{min}, Y_{max}]

The LIMIT statement defines the hard clip limits (or plotting area). The units are expressed in **millimetres** with the origin at the lower left physical limit. When the parameters aren't included, any two diagonal corner points can be digitized.

LINE TYPE id number [, length]

The LINE TYPE statement selects one of several solid or dashed line types. The range of the id number is 1 through 9; 4 is the default length.

1. Solid
2. End Point
3. Dashed
4.
5.
6.
7.
8.
9. Tic Marks

LOCATE [X_{min}, X_{max}, Y_{min}, Y_{max}]

The LOCATE statement sets the area that SHOW will fill or SCALE will map. The units are expressed in GDU's. Any two diagonal corner points can be digitized if the parameters are not included. LOCATE also invokes soft clipping at its boundaries.

LORG origin position

The LORG (label origin) statement sets the label origin position which determines where any subsequent labels are drawn relative to the current pen location. The range of the origin position is 1 through 9.

MOVE $X_{parameter}$, $Y_{parameter}$

The MOVE statement lifts the pen and moves it to the absolute X,Y coordinate.

MSCALE X_{offset}, Y_{offset}

The MSCALE statement sets **millimetres** as current user units and defines the origin. The origin is offset from the first LOCATE point (X_{min}, Y_{min}) the specified amounts in millimetres.

PDIR angle

The PDIR statement sets the angle of rotation for relative and incremental plotting. The first syntax with one parameter supplied specifies a counter-clockwise rotation from the positive X-axis in current angular units.

PEN pen number

The PEN statement specifies the pen to be used. Pen 0 (zero) specifies return all pens to their holders on the 9872A. Negative pen numbers specify "erase" on the CRT.

PENUP

The PENUP statement lifts the pen.

PLOT $X_{parameter}$, $Y_{parameter}$ [, pen control]

The PLOT statement provides absolute data plotting and pen control. The pen control defaults to plus one. The pen control parameter is interpreted as follows:

Odd	=drop pen
Even	=lift pen
Positive	=pen change after motion
Negative	=pen change before motion

PLOTTER IS [select code [, HP-IB device address],] "plotter id string"
[, step size [, # of pens, pen offset, incremental plotter id]]

The PLOTTER IS statement defines where all plotter operations will be directed. The three plotter id strings and their default select codes are:

"GRAPHICS"	13
"9872A"	7,5
"INCREMENTAL"	5

The default values for the INCREMENTAL parameters are:

step size	–	.254 mm
# of pens	–	1
pen offset	–	0 mm
incremental plotter id	–	1

PLOTTER select code [, HP-IB device address] IS OFF

The PLOTTER...IS OFF statement deactivates the plotter set to the specified select code. All plotting statements are executed but the specified plotter does not respond

PLOTTER select code [, HP-IB device address] IS ON

The PLOTTER...IS ON statement activates the specified plotting device and deactivates all others. That is, all plotting statements are directed only to the specified device. This statement **does not** set the default conditions described under the PLOTTER IS statement.

POINTER $X_{parameter}$, $Y_{parameter}$ [, cursor type]

The POINTER statement moves the cursor to the specified absolute position and can select one of two types of cursor. An even number specifies a small flashing cross; an odd number specifies full-screen crossed lines. Cursor type defaults to the full-screen crossed lines (cursor types 1).

RATIO

The RATIO function returns a value equal to the ratio of the physical dimension of the hard clip limits. That is, the X dimension divided by the Y dimension.

RPLOT $X_{parameter}$, $Y_{parameter}$ [, pen control]

The RPLOT statement allows relative plotting from the last absolute plotted point which is used as the origin. The pen control is the same as for the PLOT statement.

SCALE X_{min}, X_{max}, Y_{min}, Y_{max}

The SCALE statement sets user definable units which are mapped onto the LOCATE rectangle.

SETGU

The SETGU statement sets graphic display units (GDU's) as the current units.

SETUU

The SETUU statement sets user defined units (UDU's) as the current units.

SHOW X_{min}, X_{max}, Y_{min}, Y_{max}

The SHOW statement defines an area that is stretched or shrunk equally in X,Y directions to fit within the boundaries defined by the LOCATE statement.

UNCLIP

The UNCLIP statement sets the soft clip boundaries equal to the hard clip limits.

WHERE $X_{variable}$, $Y_{variable}$ [, pen status string variable]

The WHERE statement returns the coordinate values of the last plotted or moved-to point. Pen status information is assigned to the string variable; 0 for up, 1 for down.

Appendix E
Graphics Error Messages

ERROR	Description
110	Plotter specifications not recognized.
111	Plotter not previously specified.
112	CRT Graphics hardware not installed.
113	LIMIT specifications out of range.

Subject Index

a

Activate Plotting Device 13
Address Code, Plotter 98,99
Alphanumeric:
  Raster 2
  Mode 86
Angle 47,71
Anisotropic Scaling 21
Appendix A 93
Appendix B 97
Appendix C 103
Appendix D 111
Appendix E 119
Applications:
  Drafting 29
  Mapping 24,29
Aspect Ratio 24
AXES Statement **52,111**
AXES 44
Axes and Labeling Statements 51

b

Blank Pen 36
Box 57

c

Chapter 1 1
Chapter 2 9
Chapter 3 35
Chapter 4 51
Chapter 5 73
Chapter 6 85
Chapter Organization 8
Character Count 12
Characters, Graphics 70
Clear CRT 14
CLIP Statement **26,111**
CLIP 12,18,28,39,43,57
CLIP, Digitize 26,81
Clip, Hard Clip Limits 6
CLIP, Reflected Plots 26
CLIP, Soft Clip Boundaries 6
Copy Graphics 87
CRT Graphics Control Statements 85
CRT:
  Digitize 80
  Erase 36
  Graphics Display Area 2
  Graphics Memory 5
  ID String 10
  Line Types 37
  Memory 4
  Memory Installation 4
  Simulated Graphic Drawings 1
CSIZE Statement **68,111**
CSIZE 12,63,80
Current Units 7,31,42,43,52,75,78,80,87
CURSOR Statement **78,111**
CURSOR 63,73,75,80
Cursor Type 75
Cursor, Underline 63

d

Data Controlled Plotting 39
Deactivate Plotting Device 31
Default:
  PLOTTER IS, Id String 10
  Hard Clip Limits 15
  Incremental Values 11
  PLOTTER IS Conditions 12
Definition of Terms 7
DEG 47,71
DIGITIZE Statement **80,111**
DIGITIZE 75
Digitize:
  CLIP 26,81
  From the CRT 80
  Hard Clip Limits 16
  LIMIT 81
  LOCATE 18,81
Digitizing and Cursor Control 73
Display Area:
  Alphanumeric 3
  Graphics 2,3

Dot:
- Column 89
- Matrix 8
- Row 89

Dots, Graphics 2
Drafting 29
DRAW Statement **43,111**
DRAW 35,39,44
DUMP GRAPHICS Statement **87,112**

e

Erase 36,47
Error:
- Condition 29
- Message Sticker 2
- Message Sticker Installation 4
- Pull-out Card 4
- Graphics Messages 119

EXIT GRAPHICS Statement **86,112**
External Plotter Installation 97

f

Factory Installed Graphics 4
FRAME Statement **57,112**
FRAME 44

g

GCLEAR Statement **14,112**
GCLEAR 29,47
GDU's 4,12,18,19,26,31,37,43,44,46
General Information 1
Generation Speed:
- Curved Lines 3
- Straight Lines 3

GLOAD Pointer 88
GLOAD Statement 88,112
GRAD 47,71
GRAPHICS Statement 86,112
Graphics:
- Characters 70
- Copy 87
- Display Area 2
- Display Units (GDU's) 6,7,12
- Dots 2
- Error Message Sticker 4
- Error Messages 119
- Hardware 2
- Hardware Installation 4
- Factory Installed 4
- Memory 5
- Mode 2,86
- Options 4
- Package 4
- Raster 2
- Resolution 2
- ROM 2,93
- Select Code 5
- Updating With 4

GRID Statement **55,112**
GRID 44
GSTORE Statement **88,113**
GSTORE Pointer 88
Guidelines, Syntax 8

h

Hard Clip Limits 8,12,15,18,29,32,39,42,43,44,46,75
Hard Clip Limits:
- CRT Upper-right Values 15
- Default 15
- Digitize 16
- Origin 15
- Specifying 15

Hardware:
- Graphics 2
- Installation 4

HP 9872A Line Types 38
HP 9872A P1 and P2 16
HP 9872A Plotter Installation 97
HP 9872A,ID String 11
HP-IB:
- Device Address 10,13
- Interface 98

i

Id Number, Line Type 37
Id String:
  9872A 13
  CRT 10
  Incremental 11
Incremental:
  Plotter Installation 101
  Default Values 11
  Id String 11
Index 121
INPUT 80
Installation:
  9872A Plotter 97
  CRT Memory 4
  Error Message Sticker 4
  External Plotter 97
  Hardware 4
  Incremental Plotter 101
  ROM 4
Integer Array 88
Interface Select Code 98
Internal Printer 87
IPLOT Statement **46,113**
IPLOT 35,44,47
Isotropic Scaling 24

l

LABEL Statement **59,113**
LABEL USING Statement **59,113**
LABEL 24,44,63,71
Label, Rortate 71
LDIR Statement **71,112**
LDIR 12,63
Length, Line Type 37
LETTER Statement **63,113**
LETTER 71
LIMIT Statement **15,113**
LIMIT 12,26,28,31,39,75,93,95,119
LIMIT, Digitize 81
Line Generation 4,35
Line Generation Speed 90,3
LINE TYPE Statement **37,114**
LINE TYPE 12,36
Line Types:
  9872A 38
  CRT 37
Local Origin 46
LOCATE Statement **18,114**
LOCATE 12,21,24,26,28,39,93,95
LOCATE, Digitize 18,81
Long Side 32
LORG Statement 64 ~~71~~,**114**
LORG 12
Lower Bound 87

m

Major-tic Size 52
Manuals:
  Graphics Programming Techniques 3
  Other 1,80,88,98,101
Mapping 12,29
Mass Memory ROM 88
Memory, CRT Graphics 4,5
Metric Units 7,29
Millimetres 15
Mirror Images 93
Modes:
  Alphanumeric 2,86
  Current Units 7
  Graphics 2,86
  Metric Units 7
  Units of Measure 7
  MOVE Statement **42,115**
  MOVE 35,39,44
  MSCALE Statement **29,115**
  MSCALE 31

n

Number (#) of Pens 10
Number of Pens 29

o

Option 300 4
Option 700 4
OPTION BASE 0 88
Origin 44
Origin, Local 46
Other Example Programs 103

p

P1 and P2, 9872A 16,29
Paper Advance 14
PDIR Statement **47,115**
PDIR 44,46
PEN Statement **36,115**
PEN 12,47
Pen 73
Pen:
- Blank 36
- Control 39,44,46
- Erase 36
- Move 42
- Number 36
- Offset 10,11
- Status 35,78,84
- Status String Variable 78,84

PENUP Statement **35,115**
Physical Plotting Area 15
PLOT Statement **39,115**
PLOT 35,44
Plot, Rotate 47
Plotter Address Code 98,99
PLOTTER IS Statement **10,11,116**
PLOTTER IS 26,28,31,98,101
- Default Conditions 12
- Id String 10
- Parameters 10

PLOTTER IS OFF Statement **13,116**
PLOTTER IS ON Statement **13,116**
Plotting Area 15,18
- Physical 15
- Shape 15,18,32
- Size 15,18,32

Plotting Operations 35
Plotting Space 4
POINTER Statement **75,116**
POINTER 63,80
Pointer, GLOAD 88
Pointer, GSTORE 88
Pull-out Card, Error 4

r

RAD 47,71
RANDOMIZE 73
Raster:
- Graphics 4
- Alphanumeric 2

RATIO Function **32,117**
RATIO 32
Reflected Plots 93
- CLIP 26
- LIMIT 18
- LOCATE 18

Resolution, Graphics 2
ROM:
- Graphics 2
- Mass Memory 88
- Memory Needs 5
- Plug-in 4

Rotate:
- Label 71
- Plot 47

RPLOT Statement **44,117**
RPLOT 35,47

s

SCALE Statement **21,117**
SCALE 18,28,31,93,95
Scaling:
- Anisotropic 21
- Isotropic 24

Select Code 10,13,98
Select Code Switch 98
Set-up Operations, Typical 9
Set-up Operations 9
SETGU Statement **31,117**
SETUU Statement **31,117**
Short Side 32
SHOW Statement **24,117**
SHOW 18,21,28,31,93,95
Simulated CRT Graphic Drawings 1
Soft Clip Boundaries 6,12,18,26,39,44,46,52
- Digitize 26
- Specifying 18,26,28

Special Function Keys 80
Specifying Hard Clip Limits 15
Specifying Soft Clip Boundaries 18,26,28
Speed, Line Generation 90
Status, Pen 35
Step Size 10,11
Sticker, Error Message 4
Syntax:
- Conventions 8
- Guidelines 8
- Summary 111

t

Table of Contents iii
Typical Set-up Operations 11

u

UDU's 8,11,18,21,26,31,43,44,46
UNCLIP Statement **28,117**
UNCLIP 26
Underline Cursor 63
Units of Measure Modes 7
Units:
Current 7
GDU's 7
Metric 7
UDU's 7
Updating with Graphics 4
Upper Bound 87
User Definable Units (UDU's) 6,7,12

w

WHERE Statement **84,117**
WHERE 35

x

X axis 52
X intersection 52
X major count 52
X offset 12
X tic spacing 52

y

Y axis 52
Y intersection 52
Y major count 52
Y offset 12
Y tic spacing 52

HEWLETT hp PACKARD

SALES & SERVICE OFFICES

AFRICA, ASIA, AUSTRALIA

ANGOLA
Telectra
Empresa Técnica de Equipamentos Eléctricos, S.A.R.L.
R. Barbosa Rodrigues, 42-1°DT.
Caixa Postal, 6487
Luanda
Tel: 35515/6.
Cable: TELECTRA Luanda

AUSTRALIA
Hewlett-Packard Australia Pty. Ltd.
31-41 Joseph Street
Blackburn, Victoria 3130
P.O. Box 36
Doncaster East, Victoria 3109
Tel: 89-8351
Telex: 31-024
Cable: HEWPARD Melbourne

Hewlett-Packard Australia Pty. Ltd.
31 Bridge Street
Pymble
New South Wales, 2073
Tel: 449-6586
Telex: 21561
Cable: HEWPARD Sydney

Hewlett-Packard Australia Pty. Ltd.
153 Greenhill Road
Parkside, S.A., 5063
Tel: 272-5911
Telex: 82536 ADEL
Cable: HEWPARD ADELAID

Hewlett-Packard Australia Pty Ltd.
141 Stirling Highway
Nedlands, W.A. 6009
Tel: 86-5455
Telex: 93859 PERTH
Cable: HEWPARD PERTH

Hewlett-Packard Australia Pty. Ltd.
121 Wollongong Street
Fyshwick, A.C.T. 2809
Tel: 95-2733
Telex: 62650 Canberra
Cable: HEWPARD CANBERRA

Hewlett Packard Australia Pty. Ltd.
5th Floor
Teachers Union Building
495-499 Boundary Street
Spring Hill, 4000 Queensland
Tel: 229-1544
Cable: HEWPARD Brisbane

GUAM
Medical/Pocket Calculators Only
Guam Medical Supply, Inc.
Jay Ease Building, Room 210
P.O. Box 8947
Tamuning 96911
Tel: 646-4513
Cable: EARMED Guam

HONG KONG
Schmidt & Co.(Hong Kong) Ltd.
P.O. Box 297
Connaught Centre
39th Floor
Connaught Road, Central
Hong Kong
Tel: H-255291-5
Telex: 74766 SCHMC HX
Cable: SCHMIDTCO Hong Kong

INDIA
Blue Star Ltd.
Kasturi Buildings
Jamshedji Tata Rd
Bombay 400 020
Tel: 29 50 21
Telex: 001-2156
Cable BLUEFROST

Blue Star Ltd.
Sahas
414/2 Vir Savarkar Marg
Prabhadevi
Bombay 400 025
Tel: 45 78 87
Telex 011-4093
Cable: FROSTBLUE

Blue Star Ltd.
Band Box House
Prabhadevi
Bombay 400 025
Tel: 45 73 01
Telex: 011-3751
Cable: BLUESTAR

Blue Star Ltd
7 Hare Street
P.O. Box 506
Calcutta 700 001
Tel: 23-0131
Telex: 021-7655
Cable: BLUESTAR

Blue Star Ltd.
7th & 8th Floor
Bhandari House
91 Nehru Place
New Delhi 110024
Tel: 634770 & 635166
Telex: 031-2463
Cable: BLUESTAR

Blue Star Ltd.
Blue Star House
11/11A Magarath Road
Bangalore 560 025
Tel: 55668
Telex: 043-430
Cable: BLUESTAR

Blue Star Ltd.
Meeakshi Mandiran
xxx/1678 Mahatma Gandhi Rd.
Cochin 682 016
Tel: 32069,32161,32282
Telex: 0885-514
Cable: BLUESTAR

Blue Star Ltd.
1-1-117/1
Sarojini Devi Road
Secunderabad 500 003
Tel: 70126, 70127
Cable: BLUEFRDST
Telex: 015-459

Blue Star Ltd.
2/34 Kodambakkam High Road
Madras 600034
Tel: 82056
Telex: 041-379
Cable: BLUESTAR

INDONESIA
BERCA Indonesia P.T.
P.O. Box 496/Jkt
JLN•Abdul Muis 62
Jakarta
Tel: 40369, 49886,49255,356038
JKT 42895
Cable: BERCACON

BERCA Indonesia P.t.
63 JL. Raya Gubeng
Surabaya
Tel: 44309

ISRAEL
Electronics & Engineering Div. of Motorola Israel Ltd.
17, Kremenetski Street
P.O. Box 25016
Tel-Aviv
Tel: 38973
Telex: 33589
Cable: BASTEL Tel-Aviv

JAPAN
Yokogawa-Hewlett-Packard Ltd.
Ohashi Building
59-1 Yoyogi 1-Chome
Shibuya-ku, **Tokyo** 151
Tel: 03-370-2281/92
Telex: 232-2024YHP MARKET TOK 23-724
Cable: YHPMARKET

Yokogawa-Hewlett-Packard Ltd.
Chuo Bldg., 4th Floor
4-20, Nishinakajima 5-chome
Yodogawa-ku, Osaka-shi
Osaka 532
Tel: 06-304-6021

Yokogawa-Hewlett-Packard Ltd.
Nakamo Building
24 Kami Sasajima-cho
Nakamura-ku, **Nagoya**, 450
Tel: (052) 571-5171

Yokogawa-Hewlett-Packard Ltd.
Tanigawa Building
2-24-1 Tsuruya-cho
Kanagawa-ku
Yokohama, 221
Tel: 045-312-1252
Telex: 382-3204 YHP YOK

Yokogawa-Hewlett-Packard Ltd.
Mito Mitsu Building
105, Chome-1,San-no-maru
Mito, Ibaragi 310
Tel: 0292-25-7470

Yokogawa-Hewlett-Packard Ltd.
Inoue Building
1348-3, Asahi-cho, 1-chome
Atsugi, Kanagawa 243
Tel: 0462-24-0452

Yokogawa-Hewlett-Packard Ltd.
Kumagaya Asahi
Hachijuni Building
4th Floor
3-4, Tsukuba
Kumagaya, Saitama 360
Tel: 0485-24-6563

KENYA
Technical Engineering Services(E.A.)Ltd.,
P.O. Box 18311
Nairobi
Tel: 557726/556762
Cable: PROTON

Medical Only
International Aeradio(E.A.)Ltd.,
P.O. Box 19012
Nairobi Airport
Nairobi
Tel: 336055/56
Telex: 22201/22301
Cable: INTAERIO Nairobi

KOREA
Samsung Electronics Co., Ltd.
20th Fl. Dongbang Bldg. 250, 2-KA
C.P.O. Box 2775
Taepyung-Ro, Chung-Ku
Seoul
Tel: (23) 6811
Telex: 22575
Cable: ELEKSTAR Seoul

MALAYSIA
Teknik Mutu Sdn. Bhd.
2 Lorong 13/6A
Section 13
Petaling Jaya,**Selangor**
Tel: 54994/54916
Telex: MA 37605

Protel Engineering
P.O. Box 1917
Lot 259, Satok Road
Kuching, **Sarswak**
Tel: 2400
Cable: PROTEL ENG

MOZAMBIQUE
A.N. Goncalves, Lta.
162, 1 Apt. 14 Av. D. Luis
Caixa Postal 107
Lourenco Marques
Tel: 27091, 27114
Telex: 6-203 NEGON Mo
Cable: NEGON

NEW ZEALAND
Hewlett-Packard (N.Z.) Ltd.
P.O. Box 9443
Courtenay Place
Wellington
Tel: 877-199
Cable: HEWPACK Wellington

Hewlett-Packard (N.Z.) Ltd.
Pakuranga Professional Centre
267 Pakuranga Highway
Box 51092
Pakuranga
Tel: 569-651
Cable: HEWPACK,Auckland

Analytical/Medical Only
Medical Supplies N.Z. Ltd
Scientific Division
79 Carlton Gore Rd., Newmarket
P.O. Box 1234
Auckland
Tel: 75-289
Cable:DENTAL Auckland

Analytical/Medical Only
Medical Supplies N.Z. Ltd.
P.O. Box 1994
147-161 Tory St.
Wellington
Tel: 850-799
Telex: 3858
Cable: DENTAL, Wellington

Analytical/Medical Only
Medical Supplies N.Z. Ltd
P.O. Box 309
239 Stanmore Road
Christchurch
Tel: 892-019
Cable: DENTAL, Christchurch

Analytical/Medical Only
Medical Supplies N.Z. Ltd
303 Great King Street
P.O. Box 233
Dunedin
Tel: 88-817
Cable: DENTAL, Dunedin

NIGERIA
The Electronics Instrumentations Ltd.
N68/770 Oyo Road
Oluseun House
P.M.B. 5402
Ibadan
Tel: 61577
Telex: 31231 TEIL Nigeria
Cable: THETEIL Ibadan

The Electronics Instrumentations Ltd.
144 Agege Motor Road, Mushin
P.O. Box 6645
Lagos
Cable: THETEIL Lagos

PAKISTAN
Mushko & Company, Ltd.
Dosman Chambers
Abdullah Haroon Road
Karachi-3
Tel: 511027, 512927
Telex: 2894
Cable: COOPERATOR Karachi

Mushko & Company, Ltd.
38B, Satellite Town
Rawalpindi
Tel: 41924
Cable: FEMUS Rawalpindi

PHILIPPINES
The Online Advanced Systems Corporation
Rico House
Amorsolo cor. Herrera Str.
Legaspi Village, Makati
Metro **Manila**
Tel: 85-35-81, 85-34-91
Telex: 3274 ONLINE

RHODESIA
Field Technical Sales
45 Kelvin Road North
P.O. Box 3458
Salisbury
Tel: 705231 (5 lines)
Telex: RH 4122

SINGAPORE
Hewlett-Packard Singapore (Pte.) Ltd.
1150 Depot Road
Alexandra P.O. Box 58
Singapore 4
Tel: 270-2355
Telex: HPSG RS 21486
Cable: HEWPACK, Singapore

SOUTH AFRICA
Hewlett-Packard South Africa (Pty.) Ltd.
Private Bag Wendywood
Sandton, Transvaal 2144
Hewlett-Packard Centre
Daphne Street, Wendywood
Sandton, Transvaal 2144
Tel: 802-10408
Telex: 8-4782
Cable: HEWPACK JOHANNESBURG

Service Department
Hewlett-Packard South Africa (Pty.) Ltd.
P.O. Box 39325
Gramley, Sandton, 2018
451 Wynberg Extension 3,
Sandton, 2001
Tel: 636-8188/9
Telex: 8-2391

Hewlett-Packard South Africa (Pty.) Ltd.
P.O. Box 120
Howard Place, Cape Province, 7450
Pine Park Centre, Forest Drive,
Pinelands, Cape Province, 7405
Tel: 53-7955 thu 9
Telex: 57-0006

Service Department
Hewlett-Packard South Africa (Pty.) Ltd.
P.O. Box 37099
Overport, Durban 4067
Braby House
641 Ridge Road
Durban, 4001
Tel: 88-7478
Telex: 6-7954

TAIWAN
Hewlett-Packard Far East Ltd.,
Taiwan Branch
39 Chung Hsiao West Road
Sec. 1, 7th Floor
Taipei
Tel: 3819160-4
Cable: HEWPACK TAIPEI

Hewlett-Packard Far East Ltd.
Taiwan Branch
68-2, Chung Cheng 3rd. Road
Kaohsiung
Tel: (07) 242318-Kaohsiung

Analytical Only
San Kwang Instruments Co., Ltd.,
No. 20, Yung Sui Road
Taipei
Tel: 3715171-4 (5 lines)
Telex: 22894 SANKWANG
Cable: SANKWANG TAIPEI

TANZANIA
Medical Only
International Aeradio (E.A.), Ltd.
P.O. Box 861
Dar es Salaam
Tel: 21251 Ext. 265
Telex: 41030

THAILAND
UNIMESA Co., Ltd.
Elcom Research Building
2538 Sukumvit Ave.
Bangkok
Tel: 3932387, 3930338
Cable: UNIMESA Bangkok

UGANDA
Medical Only
International Aeradio(E.A.), Ltd.,
P.O. Box 2577
Kampala
Tel: 54388
Cable: INTAERIO Kampala

ZAMBIA
R.J. Tilbury (Zambia) Ltd.
P.O. Box 2792
Lusaka
Tel: 73793
Cable: ARJAYTEE, Lusaka

OTHER AREAS NOT LISTED, CONTACT:
Hewlett-Packard Intercontinental
3200 Hillview Ave.
Palo Alto, California 94304
Tel: (415) 493-1501
TWX: 910-373-1267
Cable: HEWPACK Palo Alto

CANADA

ALBERTA
Hewlett-Packard (Canada) Ltd.
11620A - 168th Street
EdmontonT5M 3T9
Tel: (403) 452-3670
TWX: 610-831-2431

Hewlett-Packard (Canada) Ltd.
210, 7220 Fisher St. S.E.
Calgary T2H 2H8
Tel: (403) 253-2713
Twx: 610-821-6141

BRITISH COLUMBIA
Hewlett-Packard (Canada) Ltd.
837 E. Cordova Street
Vancouver V6A 3R2
Tel: (604) 254-0531
TWX: 610-922-5059

MANITOBA
Hewlett-Packard (Canada) Ltd.
513 Century St.
St. James
Winnipeg R3H 0L8
Tel: (204) 786-7581
TWX: 610-671-3531

NOVA SCOTIA
Hewlett-Packard (Canada) Ltd.
800 Windmill Road
Dartmouth B3B 1L1
Tel: (902) 469-7820
TWX: 610-271-4482 HFX

ONTARIO
Hewlett-Packard (Canada) Ltd.
1020 Morrison Dr
Ottawa K2H 8K7
Tel: (613) 820-6483
TWX: 610-563-1636

Hewlett-Packard (Canada) Ltd.
6877 Goreway Drive
Mississauga L4V 1M8
Tel: (416) 678-9430
TWX: 610-492-4246

QUEBEC
Hewlett-Packard (Canada) Ltd.
275 Hymus Blvd.
Pointe Claire H9R 1G7
Tel: (514) 697-4232
TWX: 610-422-3022
TLX: 05-821521 HPCL

FOR CANADIAN AREAS NOT LISTED:
Contact Hewlett-Packard (Canada) Ltd. in Mississauga.

CENTRAL AND SOUTH AMERICA

ARGENTINA
Hewlett-Packard Argentina S.A.
Av. Leandro N. Alem 822 - 12
1001**Buenos Aires**
Tel: 31-6063,4,5,6 and 7
Telex: 122443 AR CIGY
Cable: HEWPACK ARG

BOLIVIA
Casa Kavlin S.A.
Calle Potosi 1130
P.O. Box 500
La Paz
Tel: 41530,53221
Telex: CWC BX 5298,ITT 3560082
Cable: KAVLIN

BRAZIL
Hewlett-Packard do Brasil I.e.C. Ltda.
Avenida Rio Negro, 980
Alphaville
06400**Barueri** SP
Tel: 429-3222

Hewlett-Packard do Brasil I.e.C. Ltda.
Rua Padre Chagas, 32
90000-**Pôrto Alegre**-RS
Tel: (0512) 22-2998, 22-5621
Cable: HEWPACK Pôrto Alegre

Hewlett-Packard do Brasil I.E.C. Ltda.
Rua Siqueira Campos, 53
Copacabana
20000-**Rio de Janeiro**
Tel: 257-80-94-DDD (021)
Telex: 391-212-I905 HEWP-BR
Cable: HEWPACK
Rio de Janeiro

CHILE
Calcagni y Metcalfe Ltda.
Alameda 580-Of. 807
Casilla 2118
Santiago, 1
Tel: 398613
Telex: 3520001 CALMET
Cable: CALMET Santiago

COLOMBIA
Instrumentacidn
Henrik A. Langebaek & Kier S.A.
Carrera 7 No. 48-75
Apartado Aéreo 6287
Bogotá, I.D.E.
Tel: 69-88-77
Cable: AARIS Bogotá
Telex: 044-400

COSTA RICA
Cientifica Costarricense S.A.
Avenida 2, Calle 5
San Pedro de Montes de Oca
Apartado 10159
San Jose
Tel: 24-38-20, 24-08-19
Telex: 2367 GALGUR CR
Cable: GALGUR

ECUADOR
Calculators Only
Computadoras y Equipos Electrdnicos
P.O. Box 6423 CCI
Eloy Alfaro #1824,3 Piso
Quito
Tel: 453482
Telex: 02-2113 Sagita Ed
Cable: Sagita-Quito

EL SALVADOR
Instrumentacion y Procesamiento Electronico de el Salvador
Bulevar de los Heroes 11-48
San Salvador
Tel: 252787

GUATEMALA
IPESA
Avenida La Reforma 3-48, Zona 9
Guatemala City
Tel: 63627, 64786
Telex: 4192 Teletro Gu

MEXICO
Hewlett-Packard Mexicana, S.A. de C.V.
Av. Periférico Sur No. 6501
Tepepan, Xochimilco
Mexico 23, D.F.
Tel: 905-678-4600

Hewlett-Packard Mexicana, S.A. de C.V.
Ave. Constitucidn No. 2184
Monterrey, N.L.
Tel: 48-71-32, 48-71-84
Telex: 038-410

NICARAGUA
Roberto Terán G.
Apartado Postal 689
Edificio Terán
Managua
Tel: 25114, 23412,23454
Cable: ROTERAN Managua

PANAMA
Electrdnico Balboa, S.A.
P.O. Box 4929
Calle Samuel Lewis
Cuidad de Panama
Tel: 64-2700
Telex: 3483103 Curunda, Canal Zone
Cable: ELECTRON Panama

PERU
Compañia Electro Médica S.A.
Los Flamencos 145
San Isidro Casilla 1030
Lima 1
Tel: 41-4325
Cable: ELMED Lima

PUERTO RICO
Hewlett-Packard Inter-Americas
Puerto Rico Branch Office
Calle 272,
No. 203 Urb. Country Club
Carolina 00924
Tel: (809) 762-7255
Telex: 345 0514

URUGUAY
Pablo Ferrando S.A.
Comercial e Industrial
Avenida Italia 2877
Casilla de Correo 370
Montevideo
Tel: 40-3102
Cable: RADIUM Montevideo

VENEZUELA
Hewlett-Packard de Venezuela C.A.
P.O. Box 50933
Caracas 105
Los Ruices Norte
3a Transversal
Edificio Segre
Caracas 107
Tel: 35-00-11 (20 lines)
Telex: 25146 HEWPACK
Cable: HEWPACK Caracas

FOR AREAS NOT LISTED, CONTACT:
Hewlett-Packard
Inter-Americas
3200 Hillview Ave.
Palo Alto, California 94304
Tel: (415) 493-1501
TWX: 910-373-1260
Cable: HEWPACK Palo Alto
Telex: 034-8300 034-8493

HP System 45 Graphics Programming Techniques

HEWLETT hp PACKARD

Part No. 09845-90050
Microfiche No. 09845-99050

Printed in U.S.A.
July 2, 1978